

МИНИСТЕРСТВО ОБРАЗОВАНИЯ И НАУКИ РФ

ГОСУДАРСТВЕННОЕ ОБРАЗОВАТЕЛЬНОЕ УЧРЕЖДЕНИЕ
ВЫСШЕГО ПРОФЕССИОНАЛЬНОГО ОБРАЗОВАНИЯ
“САМАРСКИЙ ГОСУДАРСТВЕННЫЙ АЭРОКОСМИЧЕСКИЙ
УНИВЕРСИТЕТ ИМЕНИ АКАДЕМИКА С.П. КОРОЛЕВА
(НАЦИОНАЛЬНЫЙ ИССЛЕДОВАТЕЛЬСКИЙ
УНИВЕРСИТЕТ)“

В. С. Асланов, А. С. Ледков

Нелинейная динамика

ВЫЧИСЛИТЕЛЬНЫЙ ПРАКТИКУМ

Самара 2010

УДК 531

Асланов В. С.

Нелинейная динамика: Вычислительный практикум / *Асланов В. С., Ледков А. С.* Самар. гос. аэрокосм. ун-т. Самара, 2010. 53 с.

Целью данного пособия является практическое знакомство студентов с методами нелинейной динамики. Практикум состоит из пяти частей, в каждой из которых приводится необходимый теоретический материал, пример и тридцать вариантов практических заданий для самостоятельного выполнения. Для выполнения практикума предлагается использовать математический пакет Maple.

Пособие предназначено для студентов старших курсов естественно-научных направлений, аспирантов, исследователей и может быть использовано при написании курсовых и квалификационных работ.

©Самарский государственный
аэрокосмический университет, 2010

Оглавление

1	Неподвижные точки и геометрия решений	4
1.1	Теоретическая справка	4
1.2	Необходимые Maple команды	5
1.3	Пример решения	7
1.4	Варианты заданий	12
2	Метод линеаризации	15
2.1	Теоретическая справка	15
2.2	Необходимые Maple команды	16
2.3	Пример решения	18
2.4	Варианты заданий	23
3	Отображения Пуанкаре	24
3.1	Теоретическая справка	24
3.2	Пример решения	25
3.3	Варианты заданий	29
4	Уравнение Ван дер поля	30
4.1	Теоретическая справка	30
4.2	Необходимые Maple команды	32
4.3	Пример решения	33
4.4	Варианты заданий	42
5	Метод Мельникова	45
5.1	Теоретическая справка	45
5.2	Пример решения	47
5.3	Варианты заданий	51
	Литература	52

Глава 1

Неподвижные точки и геометрия решений

Задание 1. Найти решение для линейной системы

$$\dot{x} = Ax, \quad (1.1)$$

где A - квадратная матрица размером $n \times n$ с постоянными коэффициентами, x вектор-функция размера n .

Определить неподвижные точки и изучить геометрию решений и подпространств.

1.1 Теоретическая справка

Рассмотрим линейную систему (1.1). Ее решение имеет вид

$$\phi_t(x_0) = x(x_0, t) = e^{At}x_0, \quad (1.2)$$

где e^{At} -матрица размером $n \times n$, x_0 - значение вектор-функции x в начальный момент времени $t = 0$.

Для нахождения матрицы e^{At} необходимо предварительно определить собственные значения λ_i ($i = 1..n$) и соответствующие собственные вектора v^i матрицы A . Собственные значения определяются из уравнения

$$\det(A - \lambda E) = 0, \quad (1.3)$$

где E -единичная матрица. Соответственно собственные вектора находятся из матричного уравнения

$$(A - \lambda_i E)v^i = 0. \quad (1.4)$$

В случае, если корень λ_i является кратным, то необходимо найти соответствующее значение присоединенных векторов. Сделать это можно воспользовавшись уравнением

$$(A - \lambda_i E)v^{i+1} = v^i, \quad (1.5)$$

где v^{i+1} искомый присоединенный вектор для собственного либо присоединенного вектора v^i , соответствующего собственному значению λ_i .

Матрица e^{At} ищется как

$$e^{At} = X(t)X^{-1}(0), \quad (1.6)$$

где $X(t) = [x^1(t), \dots, x^n(t)]$ - матрица фундаментальных решений, её столбцами являются решения $x^i(t)$, которые находятся как

$$x^i(t) = e^{\lambda_i t} v^i \quad (1.7)$$

для вещественных λ_i . Либо как

$$\begin{aligned} x^i(t) &= e^{\alpha t} (v^R \cos \beta t - v^t \sin \beta t), \\ x^{i+1}(t) &= e^{\alpha t} (v^R \sin \beta t + v^t \cos \beta t) \end{aligned} \quad (1.8)$$

для пары комплексных собственных значений $\lambda_i = \alpha_i + i\beta$, $\lambda_{i+1} = \alpha_i - i\beta$.

1.2 Необходимые Maple команды

Для решения задач воспользуемся математическим пакетом **Maple**. Для работы с матрицами используем библиотеку *linalg*. Перечислим некоторые команды [1]

augment(M_1, \dots, M_n) - объединяет матрицы M_i ($i \geq 2$) по горизонтали;

charpoly(M, λ) - строит характеристический полином для матрицы M с переменной λ (левая часть уравнения (1.3));

eigenvals(M) - возвращает собственные значения матрицы M ;

eigenvects(A) - возвращает список, каждый элемент которого в свою очередь является списком из трех элементов: собственного значения, его кратности и соответствующего собственного вектора;

geneqns(M, V) - генерирует из матрицы M систему алгебраических уравнений с переменными, перечисленными в V ;

inverse(M) - возвращает обратную матрицу для M ;

matrix($m, n, list$) - команда задает матрицу размером $n \times m$. Список *list* содержит записанные через запятую элементы матрицы. Перечисление элементов осуществляется строка за строкой;

mulcol(M, j, x) - умножает j -ый столбец матрицы M на скаляр x ;

multiply(M_1, M_2) - перемножает матрицы M_1 и M_2 .

transpose(M) - возвращает транспонированную матрицу для M ;

Помимо библиотеки `linalg`, воспользуемся также следующими командами:

animate($PCom, Parg, t = a..b, ptions$) - команда создает анимированный ролик. В качестве базовой команды рисования используется $PCom$, ее параметры передаются в виде списка $Parg$. При смене кадров параметр t постепенно увеличивается. При этом его нужно задействовать в $Parg$, чтобы смена кадра влияла на отображаемую картину. Число кадров можно задать в $options$ с помощью параметра $frames$.

convert($list, matrix$) - преобразовывает список $list$ в матрицу;

dsolve($\{syseq, cond\}, \{varlist\}, range = t_0..t_k, numeric$) - выполняет интегрирование системы $syseq$ с начальными условиями $cond$ на интервале времени от t_0 до t_k . При этом искомые функции перечисляются в $varlist$;

DEplot3d($eq, vars, trange, inits, options$) - команда из пакета $DEtools$, которая выводит несколько решений системы дифференциальных уравнений eq с неизвестными функциями $vars$. Интервал времени на котором ищется решение задается $trange$. Список $inits$ содержит в себе один или несколько списков, задающих начальные условия. В $option$ задаются дополнительные параметры и настройки;

plot($[x(t), y(t), t = t_0..t_k], view = [x_{min}..x_{max}, y_{min}..y_{max}]$) - строит график параметрически заданной функции, где параметр t меняется от t_0 до t_k . Значения $x_{min}, x_{max}, y_{min}, y_{max}$ задают отображаемый на графике диапазон;

PLOT($s_1, s_2, \dots, option$) - строит графические структуры, содержащие ряд объектов s_i , параметр $option$ — общие для структуры параметры. В качестве s_i могут выступать:

POINTS(p_1, p_2, \dots, p_n) — построение точек, заданных координатами p_i ;

CURVES($[p_1, p_2, \dots, p_n]$) — построение кривой по точкам с координатами p_i ;

TEXT(p, str) - печать строки str в точке с координатами p .

odeplot($RES, [x(t), y(t)], t = t_0..t_k, view = [x_{min}..x_{max}, y_{min}..y_{max}]$) - команда библиотеки $plots$, которая служит для отображения результатов интегрирования. RES - результат выполнения команды $dsolve, x(t), y(t)$ - функции из списка $varlist$ команды $dsolve$, для которых требуется построить график;

phaseportrait($eq, vars, trange, inits, options$) - команда из пакета $DEtools$ рисует решение системы дифференциальных уравнений eq с неизвестными

функциями $vars$. Интервал времени на котором ищется решение задается $trangle$. Список $inits$ содержит в себе один или несколько списков, задающих начальные условия. В $option$ задаются дополнительные параметры и настройкм. В частности, здесь можно задать диапазон координат, выводимый на график;

$solve(eq, var)$ - решение системы уравнений eq относительно неизвестных var ;

$subs(var_1 = vol_1, var_2 = vol_2, \dots, eq)$ - заменяет в выражении eq подстроки var_i на vol_i .

1.3 Пример решения

Решить задачу для матрицы коэффициентов A и начальных условий x_0

$$A = \begin{bmatrix} -2 & 1 & 3 \\ 0 & -2 & 0 \\ 1 & 0 & 0 \end{bmatrix} \quad x_0 = \begin{bmatrix} -2 \\ 1 \\ 2 \end{bmatrix}$$

Решать задачу будем в Maple.

```
> restart;
# Подключаем пакеты linalg и plots
> with(linalg):
> with(plots):
# Задаем исходную матрицу коэффициентов
> A:=matrix(3,3,[-2,1,3,0,-2,0,1,0,0]);
```

$$A := \begin{bmatrix} -2 & 1 & 3 \\ 0 & -2 & 0 \\ 1 & 0 & 0 \end{bmatrix}$$

```
# Выводим характеристический полином
> charpoly(A,lambda);
```

$$\lambda^3 + 4\lambda^2 + \lambda - 6$$

```
# Выводим собственные значения
> eigenvals(A);
```

$$1, -3, -2$$

```
# Запишем в переменную RES список, содержащий собственное
# значение, кратность и собственный вектор.
> SV:=eigenvects(A):
```

```
SV := [-2, 1, {{-2, -3, 1}}, [1, 1, {[1, 0, 1]}], [-3, 1, {{-3, 0, 1}}]
```

```
# Из переменной SV достаем координаты собственного вектора
# и преобразуем их с помощью команды convert в матрицу.
# SV[1,3] - элемент двумерного списка, содержащегося в переменной
# RES, находящийся в первой группе на третьем месте,
# то есть {{-2, -3, 1}}. Это множество, содержащее в себе список
# с координатами. Множество в Maple обозначается фигурными
# скобками, а список - квадратными. Требуемый нам список с
# координатами собственного вектора является первым и единственным
# элементом множества. Чтобы его "достать", используем конструкцию
# SV[1,3][1].
> v1:=convert(SV[1,3][1],matrix);
> v2:=convert(SV[2,3][1],matrix);
> v3:=convert(SV[3,3][1],matrix);
```

$$v1 := \begin{bmatrix} 1 \\ 0 \\ 1 \end{bmatrix} \quad v2 := \begin{bmatrix} -3 \\ 0 \\ 1 \end{bmatrix} \quad v3 := \begin{bmatrix} -2 \\ -3 \\ 1 \end{bmatrix}$$

```
# Поскольку комплексных корней нет, воспользуемся формулой (1.7)
# Умножим первый (и единственный) столбец матриц v1, v2, v3
# на экспоненту. SV[i,1] - i-е собственное значение.
> x1:=mulcol(v1,1,exp(SV[1,1]*t));
> x2:=mulcol(v2,1,exp(SV[2,1]*t));
> x3:=mulcol(v3,1,exp(SV[3,1]*t));
```

$$x1 := \begin{bmatrix} e^t \\ 0 \\ e^t \end{bmatrix} \quad x2 := \begin{bmatrix} -3e^{-3t} \\ 0 \\ e^{-3t} \end{bmatrix} \quad x3 := \begin{bmatrix} -2e^{-2t} \\ -3e^{-2t} \\ e^{-2t} \end{bmatrix}$$

```
# Зададим функцию X(x). Ниже использована довольно сложная
# конструкция. Мы соединяем в одну матрицу x1,x2,x3, которые
# в своих координатах содержат параметр t и затем заменяем
# этот параметр с помощью подстановки subs на переменную x,
# которая является аргументом функции X. Отметим, что
# конструкция # X:=(t)->augment(x1,x2,x3) будет работать
# неправильно, поскольку augment(x1,x2,x3) не содержит
# явно t (параметр t скрыт внутри x1,x2,x3) и Maple просто
# его не видит.
> X:=(x)->(subs(t=x, augment(x1,x2,x3)));
# Вычисляем матрицу по формуле (1.6)
> ETA:=multiply(X(t),inverse(X(0)));
# Зададим начальные условия
> X0:=matrix([[ -2], [1], [2]]);
# Найдем по формуле (1.2) решение для начальных условий X0
> SOL:=multiply(ETA,X0);
```

$$SOL := \begin{bmatrix} -\frac{15}{4}e^{-3t} + \frac{13}{12}e^t + \frac{2}{3}e^{-2t} \\ e^{-2t} \\ \frac{5}{4}e^{-3t} + \frac{13}{12}e^t - \frac{1}{3}e^{-2t} \end{bmatrix}$$

```
# Конечное время.
> tk:=10:
# В переменную pic1 запишем график z(x) на интервале
# времени от 0 до tk. Толщину линии установим в 3 точки.
> pic1:=plot([SOL[1,1],SOL[3,1],t=0..tk],
  view=[-5..5,-5..5],thickness=4,linestyle=dashdot):
# Чтобы удостовериться в правильности найденного решения
# проинтегрируем систему (1.1) численно. Для этого
# сформируем матрицу B, содержащую правые части
# уравнения (1.1).
> B:=multiply(A,matrix([[x(t)],[y(t)],[z(t)]]));
# В переменную eq запишем (1.1) в алгебраическом виде.
> eq:=diff(x(t),t)=B[1,1],
  diff(y(t),t)=B[2,1],
  diff(z(t),t)=B[3,1];
```

$$eq := \frac{d}{dt}x(t) = -2x(t) + y(t) + 3z(t), \quad \frac{d}{dt}y(t) = -2y(t), \quad \frac{d}{dt}z(t) = x(t)$$

```
# Численно проинтегрируем систему и запишем результат
# в переменную RES.
> RES:=dsolve({eq, x(0)=X0[1,1],y(0)=X0[2,1],z(0)=X0[3,1]}
  {x(t),y(t),z(t)},range=0..tk,numeric);
# В переменную pic2 запишем график z(x), полученный
# в результате интегрирования системы eq.
> pic2:=odeplot(RES, [x(t),z(t)], t=0..tk,
  view=[-5..5,-5..5], color=black, thickness=1);
# Покажем на одном рисунке графики pic2 и pic1 (рис.1.1).
> display (pic2,pic1);
```

Определим неподвижные точки системы (1.1) для этого решим систему

$$Ax = 0. \tag{1.9}$$

```
# Составим систему (1.9). Запишем в переменную eq2 уравнения.
# В качестве переменных используем x0,y0,z0.
> eq2:=geneqns(A, [x0,y0,z0]);
```

$$eq2 := \{x0 = 0, -2y0 = 0, -2x0 + y0 + 3z0 = 0\}$$

```
# Решим eq2 относительно [x0,y0,z0].
> solve(eq2, [x0,y0,z0]);
```




Рис. 1.1. Сравнение аналитического и численного решения

$$[[x0 = 0, y0 = 0, z0 = 0]]$$

```
# Построим фазовую траекторию в окрестности неподвижной точки
> t0:=-10:tk:=10: y00:=0:
> phaseportrait([eq], [x(t),y(t),z(t)],t=t0..tk,
  [[x(0)=0.2,y(0)=0,z(0)=0], [x(0)=2,y(0)=0,z(0)=0],
  [x(0)=4,y(0)=0,z(0)=0], [x(0)=4,y(0)=0,z(0)=2],
  [x(0)=-0.2,y(0)=0,z(0)=0], [x(0)=-2,y(0)=0,z(0)=0],
  [x(0)=-4,y(0)=0,z(0)=0], [x(0)=-4,y(0)=0,z(0)=-2],
  [x(0)=0,y(0)=0,z(0)=0.2], [x(0)=0,y(0)=0,z(0)=2],
  [x(0)=0,y(0)=0,z(0)=4], [x(0)=0,y(0)=0,z(0)=-0.2],
  [x(0)=0,y(0)=0,z(0)=-2], [x(0)=0,y(0)=0,z(0)=-4]],
  stepsize=.05, scene=[x(t),z(t)], method=classical[foreuler],
  view=[-5..5,-5..5], thickness=1, linecolor=black);
```

Чтобы лучше понять геометрию решений, создадим анимационный ролик, на котором покажем движение изображающих точек на плоскости x, z .

```
> plots[animate](DEtools[phaseportrait],
  [[eq], [x(t),y(t),z(t)],t=-1.6..tt,
  [
  [x(0)=0.2,y(0)=0,z(0)=0], [x(0)=2,y(0)=0,z(0)=0],
```




Рис. 1.2. Фазовые траектории в окрестности неподвижных точек

```

[x(0)=4,y(0)=0,z(0)=0], [x(0)=4,y(0)=0,z(0)=2],
[x(0)=-0.2,y(0)=0,z(0)=0], [x(0)=-2,y(0)=0,z(0)=0],
[x(0)=-4,y(0)=0,z(0)=0], [x(0)=-4,y(0)=0,z(0)=-2],
[x(0)=0,y(0)=0,z(0)=0.2], [x(0)=0,y(0)=0,z(0)=2],
[x(0)=0,y(0)=0,z(0)=4], [x(0)=0,y(0)=0,z(0)=-0.2],
[x(0)=0,y(0)=0,z(0)=-2], [x(0)=0,y(0)=0,z(0)=-3],
[x(0)=0,y(0)=0,z(0)=3], [x(0)=0,y(0)=0,z(0)=-1],
[x(0)=0,y(0)=0,z(0)=1], [x(0)=0,y(0)=0,z(0)=-4]],
stepsize=.05, scene=[x(t),z(t)], method=classical[foreuler],
view=[-5..5,-5..5], thickness=1, linecolor=black],
tt=-1.5..3, frames=25);
# Подключаем пакет DEtools.
> with(DEtools):
# С помощью DEplot3d создаем пространственный график, на
# котором изобразим результат решения системы уравнений eq
# с начальными условиями, перечисленными в списке..
> pic3:=DEplot3d({eq},{y(t),x(t),z(t)},t=-5..5, [
[x(0)=0,y(0)=0,z(0)=0.1], [x(0)=0,y(0)=0,z(0)=-0.1],
[x(0)=0,y(0)=-0.1,z(0)=0], [x(0)=0,y(0)=0.1,z(0)=0],
[x(0)=0.1,y(0)=0,z(0)=0], [x(0)=-0.1,y(0)=0,z(0)=0],
[x(0)=0.1,y(0)=0.1,z(0)=0], [x(0)=0.1,y(0)=-0.1,z(0)=0],
[x(0)=-0.1,y(0)=0.1,z(0)=0], [x(0)=-0.1,y(0)=-0.1,z(0)=0],
[x(0)=0.1,y(0)=0,z(0)=0.1], [x(0)=0.1,y(0)=0,z(0)=-0.1],
[x(0)=-0.1,y(0)=0,z(0)=0.1], [x(0)=-0.1,y(0)=0,z(0)=-0.1],
[y(0)=0.1,x(0)=0,z(0)=0.1], [y(0)=0.1,x(0)=0,z(0)=-0.1],
[y(0)=-0.1,x(0)=0,z(0)=0.1], [y(0)=-0.1,x(0)=0,z(0)=-0.1]
],scene=[x(t),y(t),z(t)],stepsize=0.1,
view=[-0.5..0.5,-0.5..0.5,-0.5..0.5],

```

```

linecolor=black,thickness=1):
# Преобразуем матрицы v1,v2,v3 в списки. Координаты вектора
# умножим на 0.12, чтобы вектора уместились в кубе -0.5..0.5.
> V1:=convert(transpose(v1*0.12),list)[1]:
> V2:=convert(transpose(v2*0.12),list)[1]:
> V3:=convert(transpose(v3*0.12),list)[1]:
# С помощью PLOT3D построим собственные вектора, используя
# команду CURVE для построения линий, заданных двумя точками и
# TEXT для отображения подписи с названиями векторов
>pic4:=PLOT3D(
  CURVES([-V1,V1],COLOR(RGB,1,0,0)),
  TEXT(V1*1.1,'v1',COLOR(RGB,1,0,0)),
  CURVES([-V2,V2],COLOR(RGB,1,0,0)),
  TEXT(V2*1.1,'v2',COLOR(RGB,1,0,0)),
  CURVES([-V3,V3],COLOR(RGB,1,0,0)),
  TEXT(V3*1.1,'v3',COLOR(RGB,1,0,0)),
  AXESSTYLE(BOX),VIEW(-0.5..0.5,-0.5..0.5,-0.5..0.5)):
# С помощью animate и display создадим анимацию, на которой будем
# вращать пространство с собственными векторами и фазовыми
# кривыми. Ориентацию пространства задает параметр orientation.
# Внутри этого параметра будем использовать переменную t, которая
# изменяется при смене кадра.
>plots[animate](display, [pic3,pic4,
  view=[-0.5..0.5,-0.5..0.5,-0.5..0.5],
  orientation=[t, t, 0]],t=0..360,frames=120);

```

Дополнительные материалы

Фазовые кривые в окрестности неподвижной точки (смотреть ролик)

Геометрия решений и собственные вектора (смотреть ролик)

Maple листинг (скачать файл)

1.4 Варианты заданий

$$\begin{array}{lll}
 1) A = \begin{bmatrix} 1 & 1 & 1 \\ 2 & 2 & 0 \\ 0 & 0 & 3 \end{bmatrix} & 2) A = \begin{bmatrix} 1 & 1 & 1 \\ -1 & 4 & 0 \\ 0 & 1 & 3 \end{bmatrix} & 3) A = \begin{bmatrix} 2 & 1 & 0 \\ 2 & 2 & 0 \\ -1 & 0 & 3 \end{bmatrix} \\
 4) A = \begin{bmatrix} -1 & 1 & 1 \\ 5 & 2 & 0 \\ -2 & 0 & -3 \end{bmatrix} & 5) A = \begin{bmatrix} 1 & 1 & 1 \\ 3 & 0 & -2 \\ 0 & 1 & 3 \end{bmatrix} & 6) A = \begin{bmatrix} 3 & 1 & 0 \\ 2 & 2 & 0 \\ -1 & 0 & 1 \end{bmatrix}
 \end{array}$$




Рис. 1.3. Геометрия решений и собственные вектора

$$\begin{array}{lll}
 7) A = \begin{bmatrix} 1 & 0 & 0 \\ 5 & 2 & 0 \\ -2 & 0 & -2 \end{bmatrix} & 8) A = \begin{bmatrix} -1 & 1 & 1 \\ 3 & 0 & -2 \\ 3 & 1 & 3 \end{bmatrix} & 9) A = \begin{bmatrix} 4 & 1 & 0 \\ 0 & 2 & 0 \\ -1 & 0 & 1 \end{bmatrix} \\
 10) A = \begin{bmatrix} 3 & 0 & 7 \\ 0 & 2 & 0 \\ -2 & 0 & -1 \end{bmatrix} & 11) A = \begin{bmatrix} -1 & 2 & 1 \\ 0 & 1 & -2 \\ 0 & 1 & 3 \end{bmatrix} & 12) A = \begin{bmatrix} 1 & 1 & 2 \\ 0 & -3 & 0 \\ -1 & 0 & -2 \end{bmatrix} \\
 13) A = \begin{bmatrix} 0 & 1 & 1 \\ -2 & 2 & 0 \\ 1 & 0 & 3 \end{bmatrix} & 14) A = \begin{bmatrix} 2 & 1 & 2 \\ 1 & -4 & 0 \\ 0 & 1 & 3 \end{bmatrix} & 15) A = \begin{bmatrix} -5 & 1 & 0 \\ 1 & 2 & 0 \\ -1 & 0 & 1 \end{bmatrix} \\
 16) A = \begin{bmatrix} 1 & 1 & 1 \\ 1 & 2 & 0 \\ -1 & 0 & -3 \end{bmatrix} & 17) A = \begin{bmatrix} 1 & 1 & 0 \\ 3 & -2 & -2 \\ 1 & 1 & 3 \end{bmatrix} & 18) A = \begin{bmatrix} -3 & 1 & 0 \\ 2 & 2 & 1 \\ -1 & 0 & -1 \end{bmatrix} \\
 19) A = \begin{bmatrix} 1 & 1 & 2 \\ 5 & 2 & 0 \\ 2 & 0 & -2 \end{bmatrix} & 20) A = \begin{bmatrix} -1 & 1 & 2 \\ 3 & 0 & 2 \\ -3 & 1 & 3 \end{bmatrix} & 21) A = \begin{bmatrix} 3 & 1 & 0 \\ 0 & 2 & 2 \\ -1 & 0 & 1 \end{bmatrix} \\
 22) A = \begin{bmatrix} 1 & 0 & 7 \\ 0 & 2 & 5 \\ -2 & -5 & -3 \end{bmatrix} & 23) A = \begin{bmatrix} 1 & 2 & 1 \\ 0 & -1 & -2 \\ -1 & 1 & 3 \end{bmatrix} & 24) A = \begin{bmatrix} 1 & 1 & 2 \\ 0 & 3 & 0 \\ 1 & 0 & 2 \end{bmatrix}
 \end{array}$$

$$\begin{array}{lll} 25) A = \begin{bmatrix} 3 & 0 & 1 \\ 0 & 2 & 0 \\ -2 & 1 & -1 \end{bmatrix} & 26) A = \begin{bmatrix} -1 & 2 & 1 \\ -2 & 2 & -2 \\ 0 & 1 & 3 \end{bmatrix} & 27) A = \begin{bmatrix} 1 & -1 & 2 \\ 1 & -3 & 0 \\ -1 & 0 & -2 \end{bmatrix} \\ 28) A = \begin{bmatrix} 0 & 0 & 1 \\ -3 & 2 & 0 \\ 1 & 3 & 3 \end{bmatrix} & 29) A = \begin{bmatrix} 2 & -1 & 2 \\ 1 & -1 & 0 \\ 0 & 1 & 5 \end{bmatrix} & 30) A = \begin{bmatrix} -5 & 1 & 0 \\ -1 & 2 & 0 \\ 1 & 0 & 3 \end{bmatrix} \end{array}$$

Глава 2

Метод линеаризации

Задание 2. Найти и классифицировать неподвижные точки нелинейной системы

$$\dot{x} = f(x); x \in \mathbb{R}^n, x(0) = x_0, \quad (2.1)$$

где x - вектор, $f(x)$ -вектор функция; при помощи линеаризации вблизи неподвижной точки. Найти собственные значения и собственные векторы и изобразить локальные потоки. Сравнить потоки нелинеаризованной системы с потоками исходной системы. При наличии ε рассмотреть случаи $\varepsilon < 0$, $\varepsilon = 0$, $\varepsilon > 0$

2.1 Теоретическая справка

Для определения неподвижных точек \bar{x} уравнения (2.1) нужно решить нелинейную систему

$$\dot{f}(\bar{x}) = 0. \quad (2.2)$$

После того, как неподвижные точки найдены можно перейти к изучению решений вблизи этих точек. Для этого для каждой точки \bar{x} можно записать соответствующую линейную систему:

$$\dot{\xi} = Df(\bar{x})\xi, \xi \in \mathbb{R}^n. \quad (2.3)$$

где Df - матрица Якоби

$$Df = \begin{bmatrix} \frac{\partial f_1}{\partial x_1} & \frac{\partial f_1}{\partial x_2} & \dots & \frac{\partial f_1}{\partial x_n} \\ \frac{\partial f_2}{\partial x_1} & \frac{\partial f_2}{\partial x_2} & \dots & \frac{\partial f_2}{\partial x_n} \\ \vdots & \vdots & \ddots & \vdots \\ \frac{\partial f_n}{\partial x_1} & \frac{\partial f_n}{\partial x_2} & \dots & \frac{\partial f_n}{\partial x_n} \end{bmatrix}$$

где $f(x) = [f_1(x_1, x_2, \dots, x_n), f_2(x_1, x_2, \dots, x_n), \dots, f_n(x_1, x_2, \dots, x_n)]^T$, $x = \bar{x} + \xi$.

Для двумерной систем тип особых точек можно определить с помощью собственных значений матрицы Df .

$\lambda_1 < 0, \lambda_2 < 0, Im(\lambda_i) = 0$ - устойчивый узел (рисунок 2.1 а);

$\lambda_1 > 0, \lambda_2 > 0, Im(\lambda_i) = 0$ - неустойчивый узел (рисунок 2.1 б);

$\lambda_1 > 0, \lambda_2 < 0, Im(\lambda_i) = 0$ - седло (рисунок 2.1 с);

$Re(\lambda_i) < 0, Im(\lambda_i) \neq 0$ - устойчивый фокус;

$Re(\lambda_i) > 0, Im(\lambda_i) \neq 0$ - неустойчивый фокус (рисунок 2.1 d);

$Re(\lambda_i) = 0, Im(\lambda_i) \neq 0$ - центр (рисунок 2.1 е).

Если $Df(\bar{x})$ имеет нулевую вещественную часть, то вопрос об устойчивости нельзя решать по линейному приближению и требуются дополнительные исследования.

2.2 Необходимые Maple команды

Необходимые для выполнения задния команды [2]:

dfieldplot(*eq, vars, trange, xrange, yrange, options*) - команда из пакета *DEtools*, которая методом изоклин строит фазовый портрет для двумерной системы дифференциальных уравнений *eq* с неизвестными функциями *vars*. Выводимая на график область задается интервалами *xrange, yrange*. В *option* задаются дополнительные параметры и настройкм, например параметр *arrows* задат размер стрелок, а *dirfield = [x, y]* - разбиение графика по осям *x* и *y*;

evalf(*ev*) - вычисляет значение *ev*, возвращая выражение с вещественными коэффициентами;

jacobian(*f, V*) - команда из пакета *linalg*, вычисляющая матрицу Якоби для вектор функции *f*. Вектор *V* содержит переменные по которым ищутся частные производные

Вычисление частных производных. В Maple вычисление частных производных осуществляется в несколько этапов. Пусть имеется какая-нибудь функция *f*, зависящая от двух переменных *x* и *y*, которые в свою очередь являются функциями времени.

```
> f:=a*x(t)+b*(y(t))^2;
```

$$f := ax(t) + by(t)^2$$

Для того чтобы найти частную производную нужно преобразовать *x(t), y(t)* в *x* и *y* соответственно. Сделать это можно с помощью замены *subs*:

```
> f1:=subs(x(t)=x,y(t)=y,f);
```




Рис. 2.1. Типы фазовых портретов для двумерной систем

$$f1 := ax + by^2$$

Теперь можно находить производные

```
> f1x:=diff(f1,x);
> f1y:=diff(f1,y);
```

$$f1x := a$$

$$f1y := 2by$$

После того как частные производные найдены необходимо осуществить обратную замену x и y на $x(t)$ и $y(t)$

```
> fx:=subs(x=x(t),y=y(t),f1x);
> fy:=subs(x=x(t),y=y(t),f1y);
```

$$fx := a$$

$$fy := 2by(t)$$

Конструкции выбора. Иногда бывает необходимо выполнить какое-либо действие в случае выполнения того или иного условия. В Maple для реализации таких задач используется условный оператор

```
if con then stat end if
```

где *con* - это логическое выражение, а *stat* - совокупность команд, которые выполняются если условие истинно.

2.3 Пример решения

Рассмотрим дифференциальное уравнение

$$\ddot{x} = \varepsilon \dot{x} - x + x^3.$$

Приведем его к нормальной форме

$$\begin{aligned} \dot{x} &= y, \\ \dot{y} &= \varepsilon y - x + x^3. \end{aligned}$$

Рассмотрим сначала случай, когда $\varepsilon < 0$. Примем, например, $\varepsilon = -2$. Определим неподвижные точки.

```
> restart;
# Подключаем библиотеку линейной алгебры
> with(linalg):
# Задаем систему уравнений
> eq:=diff(x(t),t)=y(t), diff(y(t),t)=epsilon*y(t)-x(t)+x(t)^3;
```

$$eq := \frac{d}{dt}x(t) = y(t), \frac{d}{dt}y(t) = \varepsilon y(t) - x(t) + x(t)^3$$

```

# Зададим в виде списка функцию правых частей
> f:=[y(t),epsilon*y(t)-x(t)+x(t)^3];
# Найдем неподвижные точки, как решение системы (2.2);
> solve([f[1]=0,f[2]=0],[x(t),y(t)]);

[[x(t) = 0,y(t) = 0],[x(t) = 1,y(t) = 0],[x(t) = -1,y(t) = 0]]

# Мы получили три неподвижные точки. Создадим переменные
# X1, X2, X3, содержащие координаты этих точек.
> X1:= [0,0]:
> X2:= [1,0]:
> X3:= [-1,0]:
# Для поиска матрицы Df необходимо вычислять частные производные.
# Заданный выше список f содержит внутри x(t) и y(t), то есть
# x и y входят в него, как функции времени. Чтобы найти частную
# производную f по x необходимо заменить в f выражение x(t) на x.
# Сделать можно с помощью команды subs, которая осуществляет
# подстановки. Создадим список f1, получаемый из f заменой x(t)
# на x и y(t) на y.
> f1:=subs(x(t)=x, y(t)=y,f);

f1 := [y, epsilon*y - x + x^3]

# Df найдем как матрицу Якоби
> Df:=jacobian(f1,[x,y]);

Df = [ 0 1
      -1 + 3x^2  epsilon ]

# Рассмотрим случай, когда epsilon=-2.
> epsilon:=-3:
# Для каждой из особых точек построим свою матрицу Df
> Df1:=subs(x=X1[1],y=X1[2],
matrix(2,2,[Df[1,1],Df[1,2],Df[2,1],Df[2,2]]));
> Df2:=subs(x=X2[1],y=X2[2],
matrix(2,2,[Df[1,1],Df[1,2],Df[2,1],Df[2,2]]));
> Df3:=subs(x=X3[1],y=X3[2],
matrix(2,2,[Df[1,1],Df[1,2],Df[2,1],Df[2,2]]));

f1 = [ 0 1 ]   Df2 = [ 0 1 ]   Df3 = [ 0 1 ]
      [-1 -3]      [ 2 -3]      [ 2 -3 ]

# Для определения типа неподвижных точек найдем
# собственные значения матриц Df. Для Df1
> eigenvals(Df1);

```

$$\frac{\sqrt{5}}{2} - \frac{3}{2}, -\frac{\sqrt{5}}{2} - \frac{3}{2}$$

```
# Команда evalf вычисляет алгебраические выражения
# и с ее помощью можно получить результат в более
# удобном виде
> evalf(eigenvals(Df1));
> evalf(eigenvals(Df2));
> evalf(eigenvals(Df3));
```

```
-0.381966012, -2.618033988
0.561552813, -3.561552813
0.561552813, -3.561552813
```

```
# Таким образом, первая точка устойчивый узел, а вторая и
# третья - седла.
```

```
# Найдем собственные вектора для неподвижных точек.
# Для первой точки получаем список, содержащий
# собственные значения, кратности и собственные вектора
> SV1:=eigenvects(Df1);
```

$$SV := \left[\left[\left[\frac{\sqrt{5}}{2} - \frac{3}{2}, 1, \left\{ \left[1, \frac{\sqrt{5}}{2} - \frac{3}{2} \right] \right\} \right], \left[-\frac{\sqrt{5}}{2} - \frac{3}{2}, 1, \left\{ \left[1, -\frac{\sqrt{5}}{2} - \frac{3}{2} \right] \right\} \right] \right]$$

```
# Извлекаем из SV список, содержащий координаты собственного
# вектора. Команда evalf вычисляет алгебраические выражения.
> V1_1:=evalf(convert(SV1[1,3][1],list));
> V1_2:=evalf(convert(SV1[2,3][1],list));
```

```
V1_1 := [0.2802517078, -0.1070466272]
V1_2 := [0.1070466269, -0.2802517076]
```

```
# Аналогично находим собственные вектора для второй и третьей точки
> SV2:=eigenvects(Df2):
> V2_1:=evalf(convert(SV2[1,3][1],list)):
> V2_2:=evalf(convert(SV2[2,3][1],list)):
> SV3:=eigenvects(Df3):
> V3_1:=evalf(convert(SV3[1,3][1],list)):
> V3_2:=evalf(convert(SV3[2,3][1],list)):
```

```
# В случае если собственные значения комплексные, eigenvects
# возвращает вектора, содержащие комплексные числа. Попытка
# нарисовать такие вектора приводит к ошибке. В таких случаях
# не будем рисовать эти вектора. Выполним проверку на
# комплексность собственного значения, и если оно содержит
```

```

# мнимую часть, запишем в собственные вектора нулевые координаты.
> if (not(Im(evalf(SV1[1,1]))=0)) then
  V1_1:=[0,0]:
  V1_2:=[0,0]:
end if:
> if (not(Im(evalf(SV2[1,1]))>0)) then
  V2_1:=[0,0]:
  V2_2:=[0,0]:
end if:
> if (not(Im(evalf(SV3[1,1]))>0)) then
  V3_1:=[0,0]:
  V3_2:=[0,0]:
end if:
# Перейдем к построению фазовых пространств.

# Подключим библиотеку DEtools.
> with(DEtools):
# Построим глобальный фазовый портрет методом изоклин.
# Запишем результат в переменную p1
> p1:=dfieldplot([eq], [x(t),y(t)],t=-2..2, x=-2..2,
  y=-2..2, arrows=medium, color=black, dirfield=[50,50]):
# В переменную p2 запишем рисунок, содержащий линии,
# проведенные вдоль собственных векторов в неподвижных
# точках
p2:=PLOT(
  CURVES([X1-V1_1,X1+V1_1],COLOR(RGB,1,0,0)),
  CURVES([X1-V1_2,X1+V1_2],COLOR(RGB,1,0,0)),
  CURVES([X2-V2_1,X2+V2_1],COLOR(RGB,1,0,0)),
  CURVES([X2-V2_2,X2+V2_2],COLOR(RGB,1,0,0)),
  CURVES([X3-V3_1,X3+V3_1],COLOR(RGB,1,0,0)),
  CURVES([X3-V3_2,X3+V3_2],COLOR(RGB,1,0,0)),
  VIEW(-2..2,-2..2),THICKNESS(1)):
# Покажем p1 и p2 на одном графике
> plots[display](p1,p2,axes=box);

```

На рисунке 2.2 показаны фазовые портреты при различных значениях ε

Вычисления собственных значений для произвольного ε показывают, что для неподвижной точки с координатами $(0, 0)$

$$\lambda_1 = \frac{1}{2} \left(\varepsilon + \sqrt{\varepsilon^2 - 4} \right), \quad \lambda_2 = \frac{1}{2} \left(\varepsilon - \sqrt{\varepsilon^2 - 4} \right); \quad (2.4)$$

а для точек $(\pm 1, 0)$

$$\lambda_1 = \frac{1}{2} \left(\varepsilon + \sqrt{\varepsilon^2 + 8} \right), \quad \lambda_2 = \frac{1}{2} \left(\varepsilon - \sqrt{\varepsilon^2 + 8} \right) \quad (2.5)$$




Рис. 2.2. Фазовая плоскость

Рассмотрим собственные значения (2.4). $|\sqrt{\varepsilon^2 - 4}| < \varepsilon$, поэтому вещественные части λ_1 и λ_2 при любом ε имеют одинаковый знак. В случае $\varepsilon > 0$ собственные значения $\lambda_{1,2} > 0$ и в неподвижная точка $(0,0)$ является неустойчивой, а в случае $\varepsilon < 0$ - устойчивым. Если $\varepsilon \in [-2; 2]$ - точка $(0,0)$ является седлом, в случае Если $|\varepsilon| > 2$ - фокусом, а при $\varepsilon = 0$ - центром.

Рассмотрим собственные значения (2.5). $|\sqrt{\varepsilon^2 + 8}| \varepsilon$, поэтому вещественные части λ_1 и λ_2 при любом ε имеют разные знаки и неподвижные точки $(\pm 1, 0)$ является седлами.

Дополнительные материалы

Фазовые кривые в окрестности неподвижной точки линейного осциллятора (смотреть ролик)

Эволюция фазовой плоскости при возрастании параметра ε (смотреть ролик)

Марле листинг (скачать файл)

2.4 Варианты заданий

$$1) \ddot{x} + \varepsilon \dot{x}^2 + x(1 + x^2) = 0,$$

$$2) \ddot{x} + \varepsilon(\dot{x} - 1)^2 + x^3 = 0,$$

$$3) \ddot{x} + \varepsilon \dot{x} + (x + 2)^3 = 0,$$

$$4) \ddot{x} + \varepsilon \dot{x}^2 + (x - 3)^2 - x^3 = 0,$$

$$5) \ddot{x} + \varepsilon \dot{x} + x^3 + x^2 = 0,$$

$$6) \ddot{x} + \varepsilon \dot{x} + x^3 - 2x^2 - x = 0,$$

$$7) \ddot{x} + \varepsilon \dot{x}(1 - x^2) + x(1 + x^2) = 0,$$

$$8) \ddot{x} + \varepsilon \dot{x}(x + 1) + x^3 + 1 = 0,$$

$$9) \ddot{x} + \varepsilon \dot{x} + \sin(x) = 0, \quad x \in [-\pi; \pi],$$

$$10) \ddot{x} + \varepsilon \dot{x}^2 - \cos(x) = 0, \quad x \in [-\pi; \pi],$$

$$11) \ddot{x} - \dot{x}^2 + \varepsilon x(1 + x^2) = 0,$$

$$12) \ddot{x} + \dot{x} + \varepsilon \sin(x) = 0, \quad x \in [-\pi; \pi],$$

$$13) \ddot{x} + \dot{x} + \varepsilon x^3 - 2x^2 = 0,$$

$$14) \ddot{x} + \dot{x}^2 + x(1 + \varepsilon x^2) = 0,$$

$$15) \ddot{x} + \dot{x} + x^3 + \varepsilon x^2 = 0,$$

$$16) \ddot{x} + \dot{x}^3 + \varepsilon x^3 - 2x^2 - x = 0,$$

$$17) \ddot{x} + \dot{x}(\varepsilon - x^2) + x(1 + x^2) = 0,$$

$$18) \ddot{x} + \dot{x} + \varepsilon x^3 \sin(x) = 0, \quad x \in [-\pi; \pi],$$

$$19) \ddot{x} + \dot{x}(x + 2)^3 + \varepsilon x = 0,$$

$$20) \ddot{x} + \dot{x}^2 + \varepsilon \cos(x) = 0, \quad x \in [-\pi; \pi],$$

$$21) \ddot{x} + \dot{x} + x^2 + \varepsilon x^3 = 0,$$

$$22) \ddot{x} + \dot{x}^2 + (x + \varepsilon)^3 = 0,$$

$$23) \ddot{x} + \varepsilon \dot{x}^2 + (x + 2)^3 = 0,$$

$$24) \ddot{x} + (x - 3)^2 - \varepsilon x^3 = 0,$$

$$25) \ddot{x} + x^3 + \varepsilon x^2 = 0,$$

$$26) \ddot{x} + \varepsilon \dot{x} + x^3 + 1 = 0,$$

$$27) \ddot{x} + \varepsilon x(1 + x^2) = 0,$$

$$28) \ddot{x} + x^3 + \varepsilon = 0,$$

$$29) \ddot{x} + \varepsilon \dot{x} \sin(x) = 0, \quad x \in [-\pi; \pi],$$

$$30) \ddot{x} + \varepsilon \dot{x}^2 + (x + 1) \sin(x) = 0, \\ x \in [-\pi; \pi].$$

Глава 3

Отображения Пуанкаре

Задание 3. Привести систему дифференциальных уравнений, описывающей движение системы с внешним возмущением, к автономной системе за счет увеличения ее размерности. Найти неподвижные точки и периодические орбиты. Построить в окрестности периодических орбит линеаризованные отображения Пуанкаре и исследовать с их помощью устойчивость орбит.

3.1 Теоретическая справка

Под *отображением* понимают динамическую систему, рассматриваемую в дискретные моменты времени. Нелинейная система и ее поток ϕ_t порождают нелинейное отображение

$$x_{n+1} = G(x_n) \quad (3.1)$$

где $G = \phi_t$ - нелинейная векторизованная функция. Если в системе имеется цикл из k различных точек $p_j = G^j(p_0)$, ($j = 0, 1, \dots, k-1$, $G^j(p_0) = G(G^{j-1}(p_0))$) и $G^k(p_0) = p_0$, то они образуют *периодическую орбиту* периода k .

Пусть γ — периодическая орбита некоторого потока ϕ в \mathbb{R}^n , порождаемого нелинейным векторным полем $f(x)$. Возьмем локальное сечение $\Sigma \subset \mathbb{R}^n$ размерности $n-1$. В каждой точке поток трансверсален гиперповерхности Σ . Обозначим (единственную) точку, где γ пересекает Σ , как p и возьмем некоторую окрестность $U \subseteq \Sigma$ точки p . *Отображение Пуанкаре*, $P : U \rightarrow \Sigma$ определяется для некоторой точки $q \in U$ как

$$P(q) = \phi_\tau(q), \quad (3.2)$$

где $\tau = \tau(q)$ — время, требующееся для того, чтобы орбита $\phi_\tau(q)$ с базой в точке q впервые вернулась на Σ .

Рассмотрим систему

$$\dot{x} = f(x, t); \quad (x, t) \in \mathbb{R}^n \times \mathbb{R}^n, \quad (3.3)$$

где функция $f(\cdot, t) = f(\cdot, t + T)$ периодична по t с периодом T . Систему (3.3) можно переписать как *автономную* за счет увеличения размерности на единицу, добавляя время в качестве явной переменной состояния:

$$\begin{aligned} \dot{x} &= f(x, \theta), \\ \dot{\theta} &= 1; \quad (x, \theta) \in \mathbb{R}^n \times S^1. \end{aligned} \quad (3.4)$$

Векторное поле f периодично по θ . Можно определить глобальное сечение

$$\Sigma = \{(x, \theta) \in \mathbb{R}^n \times S^1 \mid \theta = \theta_0\}, \quad (3.5)$$

так как все решения пересекают Σ трансверсально ввиду равенства $\dot{\theta} = 1$ в (3.4). Отображение Пуанкаре $P : \Sigma \rightarrow \Sigma$ задается формулой

$$P(x_0) = x(x_0, T + \theta_0), \quad (3.6)$$

где $x(x_0, t) = phi_t$ – решение исходной нелинейной системы с базой в точке $x(x_0, \theta_0) = x_0$.

Линеаризованные в окрестности периодических орбит отображения Пуанкаре удобно использовать для оценки устойчивости этих орбит. Если абсолютные значения всех собственных чисел λ матрицы линеаризованного отображения Пуанкаре отличны от единицы, то имеем периодическую точку гиперболического типа. Если часть собственных значений по абсолютной величине больше единицы, а часть меньше – то это точка седлового типа. Если все собственные значения по модулю меньше единицы, то это устойчивый узел или сток, и решения, которые начинаются в окрестности этой точки, сходятся к ней. Если все собственные значения по модулю больше единицы, то это неустойчивый узел. Если все собственные числа равны единице, то периодическая точка является центром.

3.2 Пример решения

Рассмотрим в качестве примера уравнение Матъе

$$\ddot{\phi} + (a^2 + b \cos t) \sin \phi = 0, \quad b \geq 0. \quad (3.7)$$

Приведем (3.7) к автономной системе

$$\begin{aligned} \dot{\phi} &= v, \\ \dot{v} &= -(a^2 + b \cos \theta) \sin \phi, \\ \dot{\theta} &= 1. \end{aligned} \quad (3.8)$$

Зададим систему (3.8) в Maple

```

> eq:=diff(phi(t),t)=v(t),
    diff(v(t),t)=-(a^2+b*cos(theta(t)))*sin(phi(t)),
    diff(theta(t),t)=1:
# Найдем неподвижные точки. Включим режим поиска
# всех решений.
> _EnvAllSolutions :=true:
# Приравняем нулю правые части исходной системы
> solve({v(t)=0,-(a^2+b*cos(theta(t)))*sin(phi(t))=0},
    {v(t),phi(t)});

```

$$\phi(t) = \Pi_{Z1}, v(t) = 0$$

```

# Здесь _Z1 означает любое целое число.
# Будем дальше рассматривать две неподвижные
# точки (0,0) и (Pi,0)
> X1:=[0,0]: X2:=[Pi,0]:

```

Заметим, что положения равновесия системы $(\phi, v) = (0, 0)$ и $(\phi, v) = (\pi, 0)$ системы в отсутствие возбуждения ($b = 0$) сохраняются и при наличии возбуждения ($b \neq 0$). Для всех b имеем периодические орбиты $(0, 0; \theta(t))$ и $(\pi, 0; \theta(t))$, где $\theta = \theta_0 + t$.

Рассмотрим сначала первую неподвижную точку. Линеаризируем систему в ее окрестности.

```

# Зададим список, содержащий правые части неавтономной системы
> f1:=[v,-(a^2+b*cos(theta))*sin(phi)];
# Найдем матрицу Якоби
> Df:=jacobian(f1,[phi,v]);

```

$$Df := \begin{bmatrix} 0 & 1 \\ -(a^2 + b \cos \theta) \cos \phi & 0 \end{bmatrix}$$

```

# Вычислим матрицу Якоби в неподвижной точке. Для этого
# с помощью subs подставим соответствующие координаты.
> Df1:=simplify(subs(phi=X1[1],v=X1[2],jacobian(f1,[phi,v])));

```

$$Df1 := \begin{bmatrix} 0 & 1 \\ -a^2 - b \cos \theta & 0 \end{bmatrix}$$

```

# Умножим Df1 на вектор переменных и с помощью subs
# заменим в результате названия переменных на
# соответствующие функции от времени
> fL1:=subs(v=v(t),phi=phi(t),theta=theta(t),
    multiply(Df1,[phi,v]));

```

$$L1 := [v(t), (-a^2 - b \cos \theta)\phi(t)]$$

```
# Составим линейную систему
> eqL1:=diff(phi(t),t)=fL1[1], diff(v(t),t)=fL1[2];
```

Исследуем с помощью полученной линейной системы

$$\begin{aligned}\dot{\phi} &= v, \\ \dot{v} &= -(a^2 + b \cos \theta)\phi, \\ \dot{\theta} &= 1.\end{aligned}$$

устойчивость периодической орбиты. Рассмотрим частный случай $b = 0$.

```
> b=0:
# Найдём аналитическое решение для линейной системы
# с начальными условиями phi0, v0.
> RES:=dsolve([eqL1,phi(0)=phi0,v(0)=v0],[phi(t),v(t)]);
```

$$RES = \left\{ \phi(t) = -\frac{-\phi_0 a \cos(at) - v_0 \sin(at)}{a}, v(t) = -\phi_0 a \sin(at) + v_0 \cos(at) \right\}$$

С помощью найденных решений зададим функции phi(t) и v(t)

```
> phi:=(t)->1/a*(phi0*a*cos(a*t)+v0*sin(a*t));
> v:=t->-phi0*a*sin(a*t)+v0*cos(a*t);
# Поскольку возмущение 2Pi периодически, определим
# отображение Пуанкаре как
P0:=[phi(2*Pi),v(2*Pi)];
```

$$P0 := \left[\frac{\phi_0 a \cos(2a\pi) + v_0 \sin(2a\pi)}{a}, -\phi_0 a \sin(2a\pi) + v_0 \cos(2a\pi) \right]$$

Найдём линеаризованный оператор DP0

```
> DP0:=jacobian(P0,[phi0,v0]);
```

$$DP0 := \begin{bmatrix} \cos(2a\pi) & \frac{\sin(2a\pi)}{a} \\ -a \sin(2a\pi) & \cos(2a\pi) \end{bmatrix}$$

Найдём собственные значения этой матрицы

```
> lambda1:=eigenvals(DP0)[1];
> lambda2:=eigenvals(DP0)[2];
```

$$\begin{aligned}\lambda_1 &:= \cos(2a\pi) + i \sin(2a\pi) \\ \lambda_2 &:= \cos(2a\pi) - i \sin(2a\pi)\end{aligned}$$

Найденные собственные значения находятся на единичной окружности, то есть орбита $(0, 0; \theta(t))$ нейтрально устойчива.

Рассмотрим случай, когда b мал. dsolve уже не позволяет найти аналитическое решение. Общее решение можно записать как

$$\begin{bmatrix} \phi \\ v \end{bmatrix} = C_1 x^1(t) + C_2 x^2(t)$$

где $x^1(t), x^2(t)$ - два линейно независимых решения. Тогда $X(t) = [x^1(t), x^2(t)]$ - матрица фундаментальных решений. Выберем такую пару независимых решений, чтобы

$$x^1(t) = \begin{bmatrix} 1 \\ 0 \end{bmatrix}, \quad x^2(t) = \begin{bmatrix} 0 \\ 1 \end{bmatrix}, \quad X(0) = I = X^{-1}(0)$$

Тогда линеаризованное отображение Пуанкаре можно получить как

$$DP_b = X(2\pi)X^{-1}(0) = X(2\pi) = \begin{bmatrix} \phi^1(2\pi) & \phi^2(2\pi) \\ v^1(2\pi) & v^2(2\pi) \end{bmatrix}$$

Можно показать, что определитель этой матрицы сохраняет свое значение и равен единице.

```
# Зададим матрицу отображения
> DPb:=matrix(2,2,[phi1,phi2,v1,v2]);
# Приравняем ее определитель единице
> eq:=det(DPb)=1;
```

$$eq := \phi_1 v_2 - \phi_2 v_1$$

```
# Выразим отсюда phi2
> phi2:=solve(eq,phi2);
# и найдем собственные значения матрицы DPb
> lambda1:=simplify(eigenvals(DPb)[1]);
> lambda2:=simplify(eigenvals(DPb)[2])
```

$$\lambda_1 = \frac{v_2}{2} + \frac{\phi_1}{2} + \frac{\sqrt{v_2^2 + 2\phi_1 v_2 + \phi_1^2 - 4}}{2}$$

$$\lambda_2 = \frac{v_2}{2} + \frac{\phi_1}{2} - \frac{\sqrt{v_2^2 + 2\phi_1 v_2 + \phi_1^2 - 4}}{2}$$

```
# Покажем, что произведение собственных значений равно 1
> expand(lambda1*lambda2);
```

1

Таким образом, собственные значения либо комплексно сопряжены и имеют ненулевые мнимые части, либо вещественны и взаимнообратны, либо кратны и равны +1 либо -1.

Дополнительные материалы

Marle листинг (скачать файл)

3.3 Варианты заданий

$$1) \ddot{\phi} + \varepsilon(\dot{\phi} - 2)^2 \cos(t) + 2\phi = 0,$$

$$2) \ddot{\phi} + \varepsilon(\dot{\phi} + 1) \sin(t) + \phi = 0,$$

$$3) \ddot{\phi} + \varepsilon\dot{\phi}^2 \cos(t) \sin \phi - \phi = 0,$$

$$4) \ddot{\phi} + \varepsilon(\dot{\phi} + 1) \cos(t) \sin^2 \phi + \phi = 0,$$

$$5) \ddot{\phi} + \varepsilon\dot{\phi} \sin(t) + \phi = 0,$$

$$6) \ddot{\phi} + \varepsilon(\dot{\phi} - 1) \cos(t) \cos^2 \phi - 3\phi = 0,$$

$$7) \ddot{\phi} + \varepsilon\dot{\phi}^2 \cos(t) \sin^2 \phi + 4\phi = 0,$$

$$8) \ddot{\phi} + \varepsilon(\dot{\phi} + 2)^2 \cos(t) \sin^2 \phi + 3\phi^2 = 0,$$

$$9) \ddot{\phi} + \varepsilon\dot{\phi} \cos(t) \sin \phi + 2\phi = 0,$$

$$10) \ddot{\phi} + \varepsilon\dot{\phi}^2 \cos(t) \cos \phi + 2\phi = 0,$$

$$11) \ddot{\phi} + \varepsilon\dot{\phi}^2 \cos(t) + 3\phi = 0,$$

$$12) \ddot{\phi} + \varepsilon\dot{\phi}^2 \cos(t) \sin^2 \phi + \phi = 0,$$

$$13) \ddot{\phi} + \varepsilon\dot{\phi} \cos(t) \sin^2 \phi + 4\phi = 0,$$

$$14) \ddot{\phi} + \varepsilon\dot{\phi}^2 \cos(t) + \phi = 0,$$

$$15) \ddot{\phi} + \varepsilon(\dot{\phi} + 2) \cos(t) \sin \phi + \phi = 0,$$

$$16) \ddot{\phi} + \varepsilon\dot{\phi}^2 \sin(t) + 5\phi = 0,$$

$$17) \ddot{\phi} + \varepsilon\dot{\phi}^2 \cos(t) + \phi = 0,$$

$$18) \ddot{\phi} + \varepsilon(\dot{\phi} - 1)^2 \cos(t) \cos \phi - \phi^2 = 0,$$

$$19) \ddot{\phi} + \varepsilon\dot{\phi}^2 \cos(t) \cos \phi + \phi = 0,$$

$$20) \ddot{\phi} + \varepsilon(\dot{\phi} - 1)^2 \cos(t) \cos^2 \phi + \phi^2 = 0,$$

$$21) \ddot{\phi} + \varepsilon\dot{\phi}^2 \cos(t) \cos^2 \phi + 3\phi = 0,$$

$$22) \ddot{\phi} + \varepsilon\dot{\phi} \cos(t) \cos \phi + \phi = 0,$$

$$23) \ddot{\phi} + \varepsilon(\dot{\phi} - 2) \cos(t) \cos \phi - \phi = 0,$$

$$24) \ddot{\phi} + \varepsilon\dot{\phi}^2 \cos(t) \sin^2 \phi + \phi = 0,$$

$$25) \ddot{\phi} + \varepsilon\dot{\phi}^2 \sin(t) + 5\phi = 0,$$

$$26) \ddot{\phi} + \varepsilon(\dot{\phi} - 2)^2 \cos(t) \sin \phi + \phi^2 = 0,$$

$$27) \ddot{\phi} + \varepsilon\dot{\phi}^2 \sin(t) + \phi = 0,$$

$$28) \ddot{\phi} + \varepsilon(\dot{\phi} - 1) \cos(t) - \phi = 0,$$

$$29) \ddot{\phi} + \varepsilon\dot{\phi}^2 \cos(t) \cos^2 \phi + 3\phi = 0,$$

$$30) \ddot{\phi} + \varepsilon\dot{\phi}^2 \cos(t) \sin \phi + 2\phi = 0.$$

Глава 4

Уравнение Ван дер Поля

Задание 4. Для уравнения Ван дер Поля

$$\ddot{x} + \alpha\phi(x)\dot{x} + x = \beta p(t), \quad (4.1)$$

где $\phi(x)$ - четная функция, отрицательная при $|x| < 1$ и положительная при $|x| > 1$, $p(t)$ - функция с периодом T , α, β - неотрицательные параметры; построить упрощенные системы для невозмущенного случая ($\beta=0$). Построить фазовый портрет невозмущенной системы. Выявить наличие предельных циклов. Для возмущенного случая провести анализ фазовых портретов при прохождении бифуркаций.

4.1 Теоретическая справка

Уравнение Ван дер Поля (4.1) является примером осциллятора с нелинейным трением, в котором энергия рассеивается при больших амплитудах и генерируется при малых амплитудах. Для таких систем типично наличие предельных циклов; они колеблются вблизи состояний, в которых приток и диссипация энергии сбалансированы, что имеет место во многих физических проблемах.

Бифуркации. Уравнение Ван дер Поля, как и многие другие из рассматриваемых в рамках нелинейной динамики системы, включают не только параметры движения, но и некоторые коэффициенты, появляющиеся при составлении уравнения. Когда эти параметры изменяются, при определенных их значениях могут произойти изменения в качественной структуре решений. Такие изменения называют *бифуркациями*, а соответствующие значения параметров — *бифуркационными значениями*. Ниже будут рассмотрены простейшие бифуркации изолированных положений равновесия и периодических орбит. Поскольку анализ таких бифуркаций обычно связан с изучением векторного поля вблизи вырожденного (бифурцирующего) положения равновесия или замкну-

той орбиты, а бифуркационные решения также ищутся вблизи этих предельных множеств, то эти бифуркации называют локальными [3].

Если

$$\dot{x} = f_\mu(x); \quad x \in \mathbb{R}^n, \mu \in \mathbb{R}^k$$

— система дифференциальных уравнений, зависящая от k -мерного параметра μ , то равновесные решения этой системы совпадают с корнями уравнения $f_\mu(x) = 0$. Как следует из теоремы о неявной функции, при изменении μ эти положения равновесия описываются гладкими функциями μ вне тех точек, в которых матрица Якоби

$$D_x f_\mu = \begin{bmatrix} \frac{\partial f_\mu^1}{\partial x_1} & \dots & \frac{\partial f_\mu^1}{\partial x_n} \\ \vdots & \ddots & \vdots \\ \frac{\partial f_\mu^n}{\partial x_1} & \dots & \frac{\partial f_\mu^n}{\partial x_n} \end{bmatrix}$$

имеет нулевое собственное значение ($f_\mu(x) = [f_\mu^1(x), f_\mu^2(x), \dots, f_\mu^n(x)]$ - вектор функции правых частей уравнения). График каждой из этих функций представляет собой некоторую ветвь положений равновесия системы. В положении равновесия (x_0, μ_0) , в котором $D_x f_\mu$ имеет нулевое собственное значение, могут сходиться несколько ветвей равновесий, такая точка (x_0, μ_0) называется точкой *бифуркации*. Если ветви равновесия показать в пространстве (x, μ) получим так называемую бифуркационную диаграмму. Бифуркационным множеством называется множество точек в пространстве μ для которых структурная устойчивость нарушается по одному из сценариев.

Метод усреднения. Рассмотрим систему вида

$$\dot{x} = \varepsilon f(x, t, \varepsilon); \quad x \in U \subseteq \mathbb{R}^n, \quad 0 \leq \varepsilon \ll 1, \quad (4.2)$$

где функция $f : \mathbb{R}^n \times \mathbb{R} \times \mathbb{R}^+$ класса C^r , $r > 2$, ограничена на ограниченных множествах и имеет по t период $T > 0$. Соответствующая *автономная усредненная система* определяется как

$$\dot{y} = \varepsilon \frac{1}{T} \int_0^T f(y, t, 0) dt = \varepsilon \bar{f}(y). \quad (4.3)$$

Теорема 4.1 (об усреднении [4]). *Существует замена переменных $x = y + \varepsilon w(y, t, \varepsilon)$ класса C^r , приводящая 4.2 к виду*

$$\dot{y} = \varepsilon \bar{f}(y) + \varepsilon^2 f_1(y, t, \varepsilon), \quad (4.4)$$

где f_1 имеет по t период T . Кроме того,

- 1) если $x(t)$ и $y(t)$ — решения систем (4.2) и (4.3) с базой в точках x_0, y_0 соответственно (при $t = 0$) и $|x_0 - y_0| = O(\varepsilon)$, то $|x(t) - y(t)| = O(\varepsilon)$ на интервале времени $t \sim 1/\varepsilon$.

- 2) Если p_0 — гиперболическая неподвижная точка для (4.3), то существует такое $\varepsilon_0 > 0$, что для всех $0 < \varepsilon \leq \varepsilon_0$ система (4.2) обладает единственной гиперболической периодической орбитой $\gamma_\varepsilon(t) = p_0 + O(\varepsilon)$ того же типа устойчивости, что и p_0 .
- 3) Если $x^s(t) \in W^s(\gamma_\varepsilon)$ — решение системы (4.2), лежащее на устойчивом многообразии гиперболической периодической орбиты $\gamma_\varepsilon(t) = p_0 + O(\varepsilon)$, $y^s(t) \in W^s(p_0)$ — решение системы (4.3), лежащее на устойчивом многообразии гиперболической неподвижной точки p_0 , причем $|x^s(0) - y^s(0)| = O(\varepsilon)$, то $|x^s(t) - y^s(t)| = O(\varepsilon)$ для всех $t \geq 0$. Аналогичный результат имеет место для решений, лежащих на неустойчивых многообразиях на интервале $t \in (-\infty, 0]$.

Из вывода (3) следует, что теорему усреднения можно использовать для аппроксимации устойчивого и неустойчивого многообразий и вообще для изучения глобальной структуры отображения Пуанкаре для системы (4.2).

Допустим, что имеется однопараметрическое семейство систем, аналогичных (4.2):

$$\dot{x} = \varepsilon f_\mu(x, t, \varepsilon); \quad \mu \in R, \quad (4.5)$$

а также соответствующее семейство усредненных систем

$$\dot{y} = \varepsilon \bar{f}_\mu(y), \quad (4.6)$$

Теорема 4.2. Если при $\mu = \mu_0$ семейство (4.6) испытывает бифуркации типа «седло-узел» или Хопфа, то для значений μ , близких к μ_0 , и достаточно малых ε отображение Пуанкаре для (4.5) испытывает бифуркацию того же типа.

Необходимо отметить, что метод усреднения непригоден для корректного описания всех бифуркаций коразмерности два (типа "вилка"), из-за тонких глобальных эффектов, связанных с гомоклинными орбитами.

4.2 Необходимые Maple команды

При выполнении практического задания нам придется задавать функции. Для создания функций в Maple используется конструкция

```
f:=(x1,x2,...,xn)->equation;
```

Здесь f - имя функции, x_1, x_2, \dots, x_n - передаваемые параметры, $equation$ - математическое выражение. Использование конструкции

```
f(x1,x2,...,xn):=equation;
```

в большинстве случаев приводит к ошибкам.

Укажем также несколько команд, для работы с аналитическими выражениями

collect (ex, x) - вынесение в выражении ex за скобки сомножителей, заданных в множестве x .

combine ($ex, form$) - объединение показателей степенных функций, тригонометрические и некоторые алгебраические преобразования в выражении ex . Требуемое преобразование задается в $form$

convert ($ex, form, arg$) - Приводит исходное выражение ex к требуемой форме, описываемой параметром $form$. arg -дополнительные аргументы.

expand (ex, arg) - расширение выражения ex . Это операция, обратная `simplify`. С помощью arg можно задать расширение отдельных фрагментов выражения.

factor (ex, arg) - разложение выражения ex на множители. Параметр arg помогает задать направление разложения.

simplify ($ex, form$) - упрощение математического выражения ex . Направление упрощений можно задать с помощью $form$.

При решении уравнений с помощью команды `solve` в результате возвращается множество выражений

{имя переменной 1=значение 1, имя переменной 2=значение 2, ... };

при этом значение не присваивается переменной. Чтобы присвоение произошло, необходимо передать результат выполнения `solve` команде `assign`.

4.3 Пример решения

Рассмотрим в качестве примера случай, когда

$$\begin{aligned}\phi(x) &= x^2 - 1; \\ p(t) &= \cos \omega t.\end{aligned}$$

Приведем уравнение Ван дер Поля к виду автономной системы

$$\begin{aligned}\dot{x} &= y - \alpha\Phi(x), \\ \dot{y} &= -x + \beta p(\theta), \\ \dot{\theta} &= 1,\end{aligned}\tag{4.7}$$

где $\Phi(x) = \int_0^x \phi(\xi) d\xi$ - нечетная функция, равная нулю при $x = 0$ и $x = \pm a$ для некоторого $a > 0$.

```

> restart;
# Задаем функции
> phi:=x->x^2-1:
> p:=t->cos(omega*t):
> Phi:=x->int(phi(xi),xi=0..x):
#Выведем Phi(x)
> Phi(x);

```

$$\frac{1}{3}x^3 - x$$

```

# Задаем систему (4.2)
> eq:=[diff(x(t),t)=y(t)-alpha*Phi(x(t)),
diff(y(t),t)=-x+beta*p(t)];

```

$$eq := \left[\frac{d}{dt}x(t) = y(t) - \alpha \left(\frac{1}{3}x(t)^3 - x(t) \right), \frac{d}{dt}y(t) = -x(t) + \beta \cos \omega t \right]$$

Двумерная система, получаемая из (4.7) при $\beta = 0$, весьма проста. Сначала предположим, что $\alpha \ll 1$ - малый параметр.

```

# Подставим в уравнения eq beta=0
> eq1:=subs(alpha=0,beta=0,eq);

```

$$eq1 := \left[\frac{d}{dt}x(t) = y(t) - \alpha \left(\frac{1}{3}x(t)^3 - x(t) \right), \frac{d}{dt}y(t) = -x(t) \right]$$

В этом случае система (4.7) является возмущением линейного осциллятора.

$$\begin{aligned} \dot{x} &= y, \\ \dot{y} &= -x, \end{aligned} \tag{4.8}$$

фазовая плоскость которого заполнена круговыми периодическими орбитами периода 2π (рисунок 4.1).

```

# Построим фазовый портрет линейного осциллятора
> dfieldplot(
  [diff(x(t),t) = y(t), diff(y(t),t) = -x(t)],
  [x(t),y(t)],t=-10..10, x=-1..1, y=-1..1,
  arrows=medium, color=black, dirfield=[30,30],
  axes=BOX);

```

Используя обычные методы теории возмущений или усреднения, можно показать, что ровно одна из этих орбит сохраняется при возмущениях. Используем для системы (4.7) обратимое преобразование

$$\begin{bmatrix} u \\ v \end{bmatrix} = \begin{bmatrix} \cos(t) & -\sin(t) \\ -\sin(t) & -\cos(t) \end{bmatrix} \begin{bmatrix} x \\ y \end{bmatrix} \tag{4.9}$$




Рис. 4.1. Фазовая плоскость системы (4.3)

```
# Подключаем библиотеку линейной алгебры.
> with(linalg):
# В переменную R1 поместим результат перемножения
# матриц в правой части (4.4). вместо x и y используем
# x1, y1.
> M:=matrix(2,2,[cos(t),-sin(t),-sin(t),-cos(t)]);
> R1:=multiply(M,[x1,y1]);
```

$$R1 := (\cos(t)x1 - \sin(t)y1, -\sin(t)x1 - \cos(t)y1)$$

```
# Разрешим систему (4.4) относительно x, y
> RES:=solve({u(t)=R1[1],v(t)=R1[2]},{x1,y1}):
# Выполним присвоение для найденных решений
> assign(RES);
# Теперь в переменных x1 и y1 содержатся найденные
# с помощью solve решения. Упростим их
> x1:=simplify(x1);
> y1:=simplify(y1);
```

$$x1 := -v(t) \sin(t) + \cos(t)u(t)$$

$$y1 := -\cos(t)v(t) - u(t) \sin(t)$$

```

# Подставим эти выражения в невозмущенную систему eq1.
> eq2:=subs(x(t)=x1,y(t)=y1,eq1):
# Разрешим eq2 относительно производных u и v
> eq3:=solve(eq2,[diff(v(t),t),diff(u(t),t)]):
# Упростим eq3 и вынесем коэффициент alpha за скобки.
> eq3:=collect(simplify(eq3[1],trig),{alpha});
# Заменем в eq3 функции u(t) и v(t) на u и v, а их
# производные на Ru и Rv соответственно.
eq4:=subs(diff(v(t),t)=Rv,diff(u(t),t)=Ru,v(t)=v,u(t)=u,eq3);
# В результате eq4 содержит список вида [Ru=..., Rv=...].
# Выполним для eq4 присвоение, после чего Ru и Rv будут
# содержать правые части невозмущенной системы.
> Ru;

$$-\frac{1}{3}\alpha \cos(t)(-3v \sin(t) \cos(t)^2 u^2 + \cos(t)^3 u^3 - v^3 \sin(t) + v^3 \sin(t) \cos(t)^2 + 3v^2 \cos(t)u - 3v^2 \cos(t)^3 u + 3v \sin(t) - 3 \cos(t)u)$$

> Rv;

$$\frac{1}{3}\alpha \sin(t)(-3v \sin(t) \cos(t)^2 u^2 + \cos(t)^3 u^3 - v^3 \sin(t) + v^3 \sin(t) \cos(t)^2 + 3v^2 \cos(t)u - 3v^2 \cos(t)^3 u + 3v \sin(t) - 3 \cos(t)u)$$

# Проведем усреднение возмущенной системы по переменной t.
> du:=simplify(1/(2*Pi)*int(Ru,t=0..2*Pi));
> dv:=simplify(1/(2*Pi)*int(Rv,t=0..2*Pi));

```

В результате получим осредненную систему:

$$\begin{aligned} du &:= -\frac{\alpha u(u^2 + v^2 - 4)}{8} \\ dv &:= -\frac{\alpha v(u^2 + v^2 - 4)}{8} \end{aligned} \quad (4.10)$$

Перейдем к переменным r, ψ с помощью замены $u = r \cos(\psi), v = r \sin(\psi)$

```

> u:=r(t)*cos(psi(t));
> v:=r(t)*sin(psi(t));
# Покажем, как x и y зависят от r и psi
> x2 := combine(-v*sin(t)+cos(t)*u,trig);
> y2 := combine(-cos(t)*v-u*sin(t),trig);

$$x2 := r(t) \cos(t + \psi(t))$$


$$y2 := -r(t) \cos(t + \psi(t))$$

# Подставим u и v в (4.5) и выполним упрощения
> eq5:=simplify([diff(u,t)=du,diff(v,t)=dv]);

```

Получим линейную относительно производных систему

$$\begin{aligned}\dot{r} \cos \psi - r \dot{\psi} \sin \psi &= -\frac{1}{8} \alpha r \cos \psi (r^2 - 4), \\ \dot{r} \sin \psi + r \dot{\psi} \cos \psi &= -\frac{1}{8} \alpha r \sin \psi (r^2 - 4).\end{aligned}$$

Разрешим ее относительно \dot{r} и $\dot{\psi}$

```
# Заменем функции от времени на переменные
> eq6:=subs(diff(r(t),t)=dr,diff(psi(t),t)=dp,r(t)=r,
           psi(t)=psi,eq5);
# Разрешим систему относительно dr, dp.
> eq7:=solve(eq6,[dr,dp]);
```

$$eq7 := \left[\left[dr = -\frac{\alpha r (r^2 - 4)}{8}, dp = 0 \right] \right]$$

Таким образом мы получили усредненную систему, которую записали в полярных координатах. Данная система точна в первом приближении и имеет погрешность $O(\alpha^2)$.

$$\begin{aligned}\dot{r} &= \frac{1}{2} \alpha r \left(1 - \frac{r^2}{4} \right) + O(\alpha^2), \\ \dot{\psi} &= 0 + O(\alpha^2).\end{aligned}\tag{4.11}$$

Пренебрегая членами $O(\alpha^2)$, можно утверждать наличие у этой системы притягивающей окружности $r = 2$, состоящей из неподвижных точек, что отражает существование однопараметрического семейства почти синусоидальных решений

$$\begin{aligned}x &= r(t) \cos(t + \psi(t)) \\ y &= -r(t) \sin(t + \psi(t))\end{aligned}\tag{4.12}$$

с медленно меняющейся амплитудой $r(t) = 2 + O(\alpha^2)$ и фазой $\psi(t) = \psi^0 + O(\alpha^2)$, где постоянная ψ^0 определяется начальными условиями.

Построим для $\alpha = 0.3$, $\omega = 0.9$ несколько фазовых траекторий (рисунок 4.2)

```
> alpha:=0.3: omega:=0.9:
> DEplot(eq1,[x(t),y(t)],t=0.0..15*Pi,
         [[x(0)=0.01,y(0)=0],[x(0)=4,y(0)=0],
          [x(0)=-4,y(0)=0],[x(0)=0,y(0)=4],[x(0)=0,y(0)=-4]],
         scene=[x(t),y(t)],stepsize=0.1,view=[-4..4,-4..4],
         linecolor=black,thickness=2,arrows=medium,
         dirfield=[30,30],thickness=1, axes=BOX);
```




Рис. 4.2. Фазовая плоскость усредненной системы при $\alpha = 0.3$, $\omega = 0.9$

Рассмотрим случай, когда $\beta \neq 0$, $\beta \ll 1$. Применим к системе (4.7) $2\pi/\omega$ периодическое преобразование.

$$\begin{bmatrix} u \\ v \end{bmatrix} = \begin{bmatrix} \cos \omega t & -\frac{1}{\omega} \sin \omega t \\ -\sin \omega t & \frac{1}{\omega} \cos \omega t \end{bmatrix} \begin{bmatrix} x \\ y \end{bmatrix}$$

```
# Создадим в Maple новый файл
> restart;
# Подключим библиотеки
> with(linalg):
> with(DEtools):
# Зададим уравнение Ван дер Поля
> phi:=x->x^2-1:
> p:=t->cos(omega*t):
> Phi:=x->int(phi(xi),xi=0..x):
> eq:=[diff(x(t),t)=y(t)-alpha*Phi(x(t)),
diff(y(t),t)=-x(t)+beta*p(t)]:
# Зададим матрицу преобразования
> M:=matrix(2,2,[cos(omega*t),
-sin(omega*t)/omega,-sin(omega*t),
-cos(omega*t)/omega]);
```

$$M := \begin{bmatrix} \cos \omega t & -\frac{\sin \omega t}{\omega} \\ -\sin \omega t & \frac{\cos \omega t}{\omega} \end{bmatrix}$$

```
# Найдем зависимость x(u,v), y(u,v)
> R1:=multiply(M, [x1,y1]):
> RES:=solve({u(t)=R1[1],v(t)=R1[2]},{x1,y1}):
> assign(RES);
> x1:=simplify(x1);
> y1:=simplify(y1);
```

$$\begin{aligned} x1 &:= -v(t) \sin(\omega t) + \cos(\omega t)u(t) \\ y1 &:= -\omega(\cos(\omega t)v(t) + u(t) \sin(\omega t)) \end{aligned}$$

```
# Действуя аналогично описанному выше способу построим
# усредненную систему.
# От переменных x,y переходим к переменным u,v.
> eq1:=subs(x(t)=x1, y(t)=y1,eq):
# Разрешаем систему относительно производных.
> eq2:=solve(eq1,[diff(v(t),t),diff(u(t),t)]):
# Упрощаем систему.
> eq3:=simplify(eq2[1],trig):
# Записываем в переменные Ru и Rv правые части системы.
> eq4:=subs(diff(v(t),t)=Rv,diff(u(t),t)=Ru,
  v(t)=v,u(t)=u,eq3):
> assign(eq4):
# Составляем усредненные уравнения
> du:=simplify(1/(2*Pi/omega)*int(Ru,t=0..2*Pi/omega));
> dv:=simplify(1/(2*Pi/omega)*int(Rv,t=0..2*Pi/omega));
```

$$\begin{aligned} du &:= -\frac{-4\alpha u\omega + \alpha u^3\omega + \alpha v^2u\omega + 4v - 4\omega^2v}{8\omega} \\ dv &:= -\frac{4\beta - 4u + \alpha v^3\omega - 4\alpha v\omega + \alpha v u^3\omega + 4v + 4\omega^2u}{8\omega} \end{aligned}$$

Введем обозначение $\sigma = (1 - \omega^2)/\alpha\omega$. Отметим, что σ является величиной порядка $O(1)$. Тогда усредненная система запишется в виде

$$\begin{aligned} \dot{u} &= \frac{\alpha}{2} \left[u - \sigma v - \frac{u}{4}(u^2 + v^2) \right] + O(\alpha^2), \\ \dot{v} &= \frac{\alpha}{2} \left[\sigma u + v - \frac{v}{4}(u^2 + v^2) \right] - \frac{\beta}{2\omega} + O(\alpha^2). \end{aligned} \quad (4.13)$$

Промасштабируем u, v с множителем 2, заменим $t \rightarrow (2/\alpha)t$ и опустим члены $O(\alpha^2)$

$$\begin{aligned} \dot{u} &= u - \sigma v - u(u^2 + v^2), \\ \dot{v} &= \sigma u + v - v(u^2 + v^2) - \gamma. \end{aligned} \quad (4.14)$$




Рис. 4.3. Бифуркационное множество для усредненного уравнения Ван дер Поля

Отображение вдоль потока усредненных уравнений (4.13) за время $2\pi/\omega$ служит аппроксимацией отображений Пуанкаре (4.1). В связи с этим исследование фазового портрета системы (4.13) или (4.14) представляет интерес. В частности, гиперболические неподвижные точки и гиперболические замкнутые орбиты системы (4.14) отвечают периодическим орбитам и, соответственно, инвариантным торам, несущим множество условно-периодических решений системы (4.1)–(4.7).

На рисунке приведено бифуркационное множество для усредненного уравнения Ван дер Поля [5]. Это множество точек, для которых поток системы структурно неустойчив. Кривые разделяют пространство параметров γ, σ области, внутри которых фазовые портреты топологически эквивалентны.

Область I - единственный сток.

Область II - два стока и седло.

Область III - единственный источник.

Область IV - сток, седло и источник.

Исследуем поведение системы при прохождении через точку A , имеющую координаты $(\sigma, \gamma) = (1/\sqrt{3}, \sqrt{8/27})$. Зафиксируем, к примеру, координату σ , а γ будем менять от $\sqrt{8/27} - 0.01$ до $\sqrt{8/27} + 0.01$.

```
# Создадим в Maple новый файл
> restart;
# Подключим библиотеки
```

```

> with(DEtools):
> with(plots):
# Зададим точку A
> sigma:=sqrt(1/3):
> gamma0:=sqrt(8/27):
# Напишем процедуру, которая в зависимости от переданного
# ей параметра gamma рисует векторное поле и несколько
# фазовых траекторий.
#
# Заголовок процедуры
> myDraw := proc(z)
# В переменную p1 запишем график нескольких фазовых траекторий
> p1:=DEplot([diff(u(t),t)=u(t)-sigma*v(t)-u(t)*(u(t)^2+v(t)^2),
             diff(v(t),t)=sigma*u(t)+v(t)-v(t)*(u(t)^2+v(t)^2)-z],
             [u(t),v(t)],t=-150..150,
             [[u(0)=0.1,v(0)=0],[u(0)=5,v(0)=5],[u(0)=-5,v(0)=5],
             [u(0)=-5,v(0)=-5],[u(0)=5,v(0)=-5],[u(0)=0.9,v(0)=0],
             [u(0)=1.1,v(0)=0],[u(0)=0.7,v(0)=0.4],
             [u(0)=0.8,v(0)=0.4]], scene=[u(t),v(t)],stepsize=0.1,
             view=[-2..2,-2..2], linecolor=black,thickness=1,
             arrows=none):
# В переменную pp2 запишем график векторного поля
> pp2:=dfieldplot([
             diff(u(t),t)=u(t)-sigma*v(t)-u(t)*(u(t)^2+v(t)^2),
             diff(v(t),t)=sigma*u(t)+v(t)-v(t)*(u(t)^2+v(t)^2)-z],
             [u(t),v(t)],t=-10..10, u=-2..2, v=-2..2,
             arrows=medium,color=red,dirfield=[30,30],axes=BOX):
# Совместим графики и покажем результат
> display(pp1,pp2,axes=BOX, view=[-2..2,-2..2]);
> end proc:
# Создадим анимацию. Пусть при смене кадра изменяется параметр
# gamma. Причем его начальное значение равно gamma0-0.1,
# а конечное gamma0+0.1.
>plots[animate](myDraw, [gamma2],
                gamma2=gamma0-0.1..gamma0+0.1, frames=31);

```

Проанализируем полученный результат. При $\gamma < \gamma_0$ система находится в состоянии соответствующем третьей зоне на рисунке 4.3, когда на фазовом портрете существует единственная неподвижная точка - источник и устойчивый предельный цикл (рисунок 4.4 a). При переходе через $\gamma = \gamma_0 = \sqrt{8/27}$ происходит бифуркация. Этому положению соответствует единственная негиперболическая неподвижная точка с нулевым и отрицательным собственным значением - вырожденный узел (рисунок 4.4 b). При $\gamma > \gamma_0$ система находится в первой области области бифуркационного множества 4.3 и имеет единственную

неподвижную точку - сток (рисунок 4.4 с).

Дополнительные материалы

Эволюция фазового портрета уравнения Ван дер Поля при прохождении через точку бифуркации (смотреть ролик)

Marle листинг часть 1 (скачать файл)

Marle листинг часть 2 (скачать файл)

Marle листинг часть 3 (скачать файл)

4.4 Варианты заданий

Задание 1. Для уравнения Ван дер Поля (4.1) получить усредненную систему для невозмущенного случая ($\beta=0$). Построить фазовый портрет этой системы. Для возмущенной системы построить бифуркационное множество, аналогичное представленному на рисунке 4.3. В качестве параметров принять

- | | |
|--|---|
| 1) $\phi = x^2 + x - 1; p = \cos \omega t$ | 2) $\phi = x^2 + x - 2; p = \cos \omega t$ |
| 3) $\phi = x^2 + x - 1/2; p = \cos \omega t$ | 4) $\phi = x^2 - x - 1; p = \cos \omega t$ |
| 5) $\phi = x^2 - x - 2; p = \cos \omega t$ | 6) $\phi = x^2 - x - 1/2; p = \cos \omega t$ |
| 7) $\phi = 2x^2 - x - 1; p = \cos \omega t$ | 8) $\phi = 2x^2 - x - 2; p = \cos \omega t$ |
| 9) $\phi = 2x^2 - x - 3/2; p = \cos \omega t$ | 10) $\phi = 2x^2 - x/2 - 1; p = \cos \omega t$ |
| 11) $\phi = 2x^2 - x/3 - 1; p = \cos \omega t$ | 12) $\phi = x^2/3 - x - 1; p = \cos \omega t$ |
| 13) $\phi = 2x^2 - 2; p = \cos \omega t$ | 14) $\phi = 3x^2 - x - 1; p = \cos \omega t$ |
| 15) $\phi = 3x^2 - x/2 - 2; p = \cos \omega t$ | 16) $\phi = x^2 + x - 2; p = \sin \omega t$ |
| 17) $\phi = x^2 + x/3 - 1; p = \sin \omega t$ | 18) $\phi = x^2 - x - 1/4; p = \sin \omega t$ |
| 19) $\phi = x^2 - x/5 - 2; p = \sin \omega t$ | 20) $\phi = x^2 - x - 1/2; p = \sin \omega t$ |
| 21) $\phi = 3x^2 - x - 1; p = \sin \omega t$ | 22) $\phi = 2x^2 - x - 2; p = \sin \omega t$ |
| 23) $\phi = 2x^2 - x - 3/4; p = \sin \omega t$ | 24) $\phi = 2x^2 - x/2 - 1; p = \sin \omega t$ |
| 25) $\phi = 2x^2 - x/3 - 1; p = \sin \omega t$ | 26) $\phi = x^2/3 - x - 1/2; p = \sin \omega t$ |
| 27) $\phi = 2x^2 - 2; p = \sin \omega t$ | 28) $\phi = 3x^2 - x - 1; p = \sin \omega t$ |
| 29) $\phi = 3x^2 - x/2 - 2; p = \sin \omega t$ | 30) $\phi = 3x^2 - x - 1; p = \sin \omega t$ |

Задание 2. Провести анализ фазовых портретов усредненной системы (4.14) при

$$\begin{aligned}\phi(x) &= x^2 - 1; \\ p(t) &= \cos \omega t,\end{aligned}$$

когда параметры σ и γ изменяются как описано ниже. С помощью анимации показать эволюцию фазового портрета.

Параметры σ и γ меняются по линейному закону

- | | |
|---|---|
| 1) $\sigma = 0.1, \gamma \in [0.25, 0.5]$ | 2) $\sigma = 0.25, \gamma \in [0.3, 0.5]$ |
| 3) $\sigma = 0.25, \gamma \in [0.1, 0.3]$ | 4) $\sigma = 0.5, \gamma \in [0.3, 0.5]$ |




Рис. 4.4. Фазовые портреты уравнения Ван дер Поля

5) $\sigma = 0.5, \gamma \in [0.5, 0.6]$

6) $\sigma = 0.6, \gamma \in [0.4, 0.7]$

7) $\sigma = 0.75, \gamma \in [0.5, 0.75]$

8) $\sigma = 0.9, \gamma \in [0.5, 0.8]$

9) $\sigma = 0.4, \gamma \in [0.25, 0.5]$

10) $\sigma = 0.8, \gamma \in [0.6, 0.8]$

11) $\gamma = 0.2, \sigma \in [0.1, 0.5]$

12) $\gamma = 0.25, \sigma \in [0.5, 0.7]$

13) $\gamma = 0.5, \sigma \in [0.25, 0.5]$

14) $\gamma = 0.5, \sigma \in [0.5, 0.75]$

15) $\gamma = 0.6, \sigma \in [0.5, 0.75]$

16) $\gamma = 0.65, \sigma \in [0.5, 0.8]$

17) $\gamma = 0.7, \sigma \in [0.5, 0.8]$

18) $\gamma = 0.75, \sigma \in [0.5, 0.8]$

19) $\gamma = 0.8, \sigma \in [0.6, 0.9]$

20) $\gamma = 0.5, \sigma \in [0.25, 0.75]$

Параметры σ и γ движутся против часовой стрелки по окружности с центром в точке $M = (\sigma_0, \gamma_0)$, радиус окружности R .

21) $\sigma_0 = 1/\sqrt{3}, \gamma_0 = \sqrt{8/27}, R = 0.05$

22) $\sigma_0 = 1/\sqrt{3}, \gamma_0 = \sqrt{8/27}, R = 0.1$

23) $\sigma_0 = 1/\sqrt{3}, \gamma_0 = \sqrt{8/27}, R = 0.2$

24) $\sigma_0 = 1/\sqrt{3}, \gamma_0 = \sqrt{8/27}, R = 0.3$

25) $\sigma_0 = 1/\sqrt{3}, \gamma_0 = \sqrt{8/27}, R = 0.4$

26) $\sigma_0 = 1/2, \gamma_0 = 1/2, R = 0.05$

27) $\sigma_0 = 1/2, \gamma_0 = 1/2, R = 0.1$

28) $\sigma_0 = 1/2, \gamma_0 = 1/2, R = 0.15$

29) $\sigma_0 = 1/2, \gamma_0 = 1/2, R = 0.2$

30) $\sigma_0 = 0, \gamma_0 = 0, R = 0.5$ (только первый квадрант)

Глава 5

Метод Мельникова

Задание 5. Для данной двумерной периодической по времени системы

$$\dot{x} = f(x) + \varepsilon g(x, t); \quad x = \begin{bmatrix} u \\ v \end{bmatrix} \in \mathbb{R}^2, \quad (5.1)$$

получить функцию Мельникова с помощью которой проверить пересекаются ли устойчивое и неустойчиво многообразия $W^s(p_\varepsilon)$ и $W^u(p_\varepsilon)$. Для случая пересечений многообразий построить несколько фазовых кривых, подтверждающих переход фазовых траекторий из одной области в другую.

5.1 Теоретическая справка

Рассмотрим систему (5.1), где

$$f = \begin{bmatrix} f_1(x) \\ f_2(x) \end{bmatrix}, \quad g = \begin{bmatrix} g_1(x, t) \\ g_2(x, t) \end{bmatrix}$$

достаточно гладкие (класса C^r , $r \geq 2$) и ограниченные на ограниченных множествах, а g имеет по t период T . Мы полагаем для простоты, что невозмущенная система гамильтонова:

$$f_1 = \frac{\partial H}{\partial v}, \quad f_2 = -\frac{\partial H}{\partial u}.$$

Пусть на систему наложены также следующие ограничения:

- 1) При $\varepsilon = 0$ система (5.1) обладает орбитой $q^0(t)$, гомоклинической к гиперболической седловой точке p_0 .
- 2) Положим $\Gamma^0 = \{q^0(t) | t \in \mathbb{R}\} \cap \{p_0\}$. Внутренность Γ^0 заполнена непрерывным семейством периодических орбит $q^\alpha(t)$, $\alpha \in (-1, 0)$. Полагая $d(x, \Gamma^0) = \inf_{q \in \Gamma^0} |x - q|$, имеем $\limsup_{\alpha \rightarrow 0} \limsup_{t \in \mathbb{R}} d(q^\alpha(t), \Gamma^0) = 0$.

3) Пусть $h_\alpha = H(q^\alpha(t))$, а T_α — период орбиты $q^\alpha(t)$. Тогда T_α — дифференцируемая функция от h_α , причем внутри Γ^0 имеем $\frac{dT_\alpha}{dh_\alpha} > 0$.

Лемма. При сделанных выше предположениях, для достаточно малых ε система (5.1) имеет единственную гиперболическую орбиту $\gamma_\varepsilon^0(t) = p_0 + O(\varepsilon)$. Соответственно, отображение Пуанкаре $P_\varepsilon^{t_0}$ имеет единственную гиперболическую седловую точку $p_\varepsilon^{t_0} = p_0 + O(\varepsilon)$.

Теорема 5.1. Если функция $M(t_0)$

$$M(t_0) = \int_{-\infty}^{\infty} f(q^0(t-t_0)) \wedge g(q^0(t-t_0), t) dt \quad (5.2)$$

имеет простые нули и не зависит от ε , то $W^u(p_\varepsilon^{t_0})$ и $W^s(p_\varepsilon^{t_0})$ пересекаются трансверсально. Если значения $M(t_0)$ отделены от нуля, то $W^u(p_\varepsilon^{t_0}) \cap W^s(p_\varepsilon^{t_0}) = \emptyset$.

Внешнее произведение в (5.2) определяется как

$$a \wedge b = a_1 b_2 - a_2 b_1.$$

Замена переменных $t \rightarrow t + t_0$ приводит интеграл Мельникова (5.2) к виду более удобному для расчетов:

$$M(t_0) = \int_{-\infty}^{\infty} f(q^0(t)) \wedge g(q^0(t), t + t_0) dt \quad (5.3)$$

Важность данной теоремы обусловлена возможностью проверки существования трансверсальных гомоклинических орбит для конкретных дифференциальных уравнений. Из присутствия таких орбит следует, что некоторая итерация $P_\varepsilon^{t_0}$ отображения Пуанкаре имеет инвариантное гиперболическое множество — подкову Смейла, которое содержит множество (неустойчивых) периодических орбит, несчетное множество ограниченных непериодических орбит и плотную орбиту. Чувствительность к выбору начальных условий, которую она придает потоку для данного дифференциального уравнения, представляет большой практический интерес.

Теорема 5.2. Рассмотрим параметризованное семейство $\dot{x} = f(x) + \varepsilon g(x, t; \mu)$, $\mu \in \mathbb{R}$ для которого выполнены гипотезы 1-3. Допустим, что функция Мельникова $M(t_0, \mu)$ имеет (квадратичный) корень: $M(\tau, \mu_b) = \left(\frac{\partial M}{\partial t_0}\right)(\tau, \mu_b) = 0$, но $\left(\frac{\partial^2 M}{\partial t_0^2}\right)(\tau, \mu_b) \neq 0$ и $\left(\frac{\partial M}{\partial \mu}\right)(\tau, \mu_b) \neq 0$. Тогда $\mu_B = \mu_b + O(\varepsilon)$ — бифуркационное значение, для которого в данном семействе систем имеют место квадратичные гомоклинические касания.

5.2 Пример решения

Рассмотрим уравнение Дуффинга с отрицательной линейной жесткостью и слабым синусоидальным возбуждением и демпфированием. Записывая это уравнение в виде (5.1), получим

$$\begin{aligned}\dot{u} &= v, \\ \dot{v} &= u - u^3 + \varepsilon(\gamma \cos \omega t - \delta v),\end{aligned}\tag{5.4}$$

где амплитуда силы γ , частота ω и коэффициент демпфирования δ представляют собой изменяемые параметры, а ε — малый (масштабирующий) параметр.

Гамильтониан системы (5.4) имеет вид

$$H = \frac{v^2}{2} - \frac{u^2}{2} + \frac{u^4}{4}$$

в чем нетрудно убедиться

```
> restart;
> with(linalg):
# Задаем гамильтониан
> H:=v^2/2-u^2/2+u^4/4;
# Убедимся, что он соответствует исследуемой системе
> du=diff(H,v);
> dv=-diff(H,u);
```

$$\begin{aligned}du &= v \\ dv &= u - u^3\end{aligned}$$

```
# Зададим список для правых частей уравнения и для возмущений
> f:=[diff(H,v),-diff(H,u)]:
> g:=[0,gamma1*cos(omega*t)-delta*v]:
# Найдем неподвижные точки невозмущенной системы
> RES:=solve(f,{u,v});
```

$$RES := \{u = 0, v = 0\}, \{u = 1, v = 0\}, \{u = -1, v = 0\}$$

```
# Определим седловую точку, для этого составим матрицу
# линеаризованного отображения Пуанкаре и найдем ее
# собственные значения
> Df:=jacobian(f,[u,v]);
> ev:=eigenvals(Df);
# Выведем координаты неподвижной точки и соответствующие
# собственные значения.
> RES[1],subs(RES[1],evalf([ev]));
> RES[2],subs(RES[2],evalf([ev]));
> RES[3],subs(RES[3],evalf([ev]));
```

$$\begin{aligned} &\{u = 0, v = 0\}, [1., -1] \\ &\{u = 1, v = 0\}, [1.414213562 I, -1.414213562 I] \\ &\{u = -1, v = 0\}, [1.414213562 I, -1.414213562 I] \end{aligned}$$

Видно, что седловая точка имеет координаты $(u, v) = (0, 0)$

Для рассматриваемой системы полная энергия является первым интегралом, то есть

$$H = \text{const} = h$$

Подставляя координаты седловой точки в уравнение для Гамильтониана получим $h = 0$. Таким образом гомоклинические траектории, соответствующие движению по сепаратрисе описываются дифференциальным уравнением

$$H = \frac{\dot{u}^2}{2} - \frac{u^2}{2} + \frac{u^4}{4} = 0$$

Проинтегрируем аналитически последнее уравнение

> dsolve(diff(u(t), t)^2/2 - u(t)^2/2 + u(t)^4/4, u(t));

$$u(t) = \sqrt{2}, u(t) = -\sqrt{2}, u(t) = \frac{\sqrt{2 \tanh(-t + _C1)^2 - 2}}{\tanh(-t + _C1)}$$

$$u(t) = -\frac{\sqrt{2 \tanh(-t + _C1)^2 - 2}}{\tanh(-t + _C1)}$$

Решения $u(t) = \pm\sqrt{2}$ соответствуют точкам $(u, v) = (\pm\sqrt{2}, 0)$ для которых $H=0$. Другие два решения описывают невозмущенные гомоклинические траектории $q^0(t)$

Для определения константы $_C1$ подставим в $u(t)$ координаты неподвижной точки.

Зададим функцию $u(t, _C1)$, используя полученные решения.

Перейдем от \tanh к \sinh .

> u:=(t, _C1)->convert((2*tanh(-t+_C1)^2-2)^(1/2)/
tanh(-t+_C1),sinh);

Найдем константу $_C1$ из условия, что при $t=0$ $u=0$

> RES2:=solve(u(0,x)=0,x);

$$RES2 := 1/2I\pi, -1/2I\pi$$

Подстановка любого из найденных значений приводит к одному и

тому же результату. Чтобы убедиться в этом, найдем разность

> u(t, RES2[1]) - u(t, RES2[2]);

0




Рис. 5.1. Гомоклинические траектории невозмущенной системы

```

# Используя найденную константу создадим функцию для левой и правой
# ветви сепаратрисы
> u1:=t->u(t,RES2[1]):
> u2:=t->-u(t,RES2[1]):
# Зависимости v(t) найдем дифференцируя u(t)
> v1:=t->subs(q=t,factor(diff(u1(q),q)));
> v2:=t->subs(q=t,factor(diff(u2(q),q)));
# Построим сепаратрису с помощью найденных решений
plot([[u1(t),diff(u1(t),t),t=0..10],
      [u2(t),diff(u2(t),t),t=0..10],
      [u1(t),diff(u1(t),t),t=-10..0],
      [u2(t),diff(u2(t),t),t=-10..0]]);

# Приступим к построению функции Мельникова. Рассмотрим одну из
# найденных гомоклинических траекторий.
# Подставим в списки f и g решения u1(t), v1(t)
> fq0:=subs(u=u1(t),v=v1(t),f):
> gq0:=subs(t=t+t0,u=u1(t),v=v1(t),g):
# Тогда согласно (5.3) подинтегральное выражение имеет вид
> Mint:=factor(fq0[1]*gq0[2]-fq0[2]*gq0[1]):
# Для удобства в качестве параметров функции М укажем не
# только t0, но и другие параметры системы
> M:=(x0,x1,x2,x3)->int(subs(t0=x0,gamma1=x1,delta=x2,omega=x3,Mint),t=-15..15);
# Здесь мы в качестве пределов интегрирования использовали пределы
# -15..15. Экспериментальным путем было установлено, что такие пределы
# позволяют численно вычислить полученный интеграл с точностью до четырех
# знаков после запятой

```

В качестве примера построим зависимость $M(t_0)$ для $\gamma = 0.4$, $\delta = 0.1$, $\omega = 1$ (рисунок 5.2).

```
plot(M(t0,0.4,0.1,1),t0=-5..5,numpoints=2)
```




Рис. 5.2. Функция Мельникова для левой гомоклинической траектории

График пересекает ось абсцисс, значит $M(t_0)$ имеет простые нули и $W^u(p_\varepsilon^{t_0})$ и $W^s(p_\varepsilon^{t_0})$ пересекаются трансверсально.

Построим фазовую траекторию, иллюстрирующую пересечение гомоклинических траекторий и переход из одной области движения в другую (рисунок 5.3).

```
# В переменную P0 запишем график невозмущенной сепаратрисы
> P0:=plot([[u1(t),diff(u1(t),t),t=0..10],
           [u2(t),diff(u2(t),t),t=0..10],
           [u1(t),diff(u1(t),t),t=-10..0],
           [u2(t),diff(u2(t),t),t=-10..0]]),
           color=[red,red,red,red]):

# Расчитываем фазовую траекторию и записываем результат
# в переменную P1.
> P1:=DEplot([diff(x(t),t) = y(t),
             diff(y(t),t) = x(t)-x(t)^3+0.4*cos(t)-0.25*y(t)],
             [x(t),y(t)],t=0..30,[[x(0)=0.3,y(0)=0]],
             scene=[x(t),y(t)],stepsize=0.1,
             view=[-1.5..1.5,-1.5..1.5], linecolor=black,
             thickness=1, axes=BOX, maxfun=0, arrows=none):

# Строим P0 и P1 на одном графике
> display(P0,P1);
```




Рис. 5.3. Переход фазовой траектории из одной области движения в другую

Дополнительные материалы

Ниже приведены ссылки на несколько видеороликов, иллюстрирующие поведение системы и перетекание траекторий из одной области невозмущенной системы в другую при различных соотношениях параметров. Во всех примерах $\omega = 1$. Серым цветом показаны траектории, которые изначально находились во внешней области, зеленым - в левой внутренней, коричневым - в правой внутренней. Красным цветом показана сепаратриса невозмущенной системы.

- Невозмущенная система $\gamma = 0$, $\delta = 0$ (смотреть ролик).
- Определяющее влияние оказывает диссипативная сила $\gamma = 0.01$, $\delta = 0.2$ (смотреть ролик).
- Определяющее влияние оказывает периодическое возмущение $\gamma = 0.1$, $\delta = 0.01$ (смотреть ролик).
- Периодическое возмущение и диссипативная сила сопоставимы $\gamma = 0.1$, $\delta = 0.1$ (смотреть ролик).

Marle листинг (скачать файл)

5.3 Варианты заданий

Исследовать аналогичным образом систему
Для вариантов 1-10:

$$\begin{aligned}\dot{\theta} &= v, \\ \dot{v} &= -\sin \theta + \varepsilon(\alpha + \gamma \cos \omega t);\end{aligned}$$

Для вариантов 11-20:

$$\begin{aligned}\dot{\theta} &= v, \\ \dot{v} &= -\sin\theta + \alpha\theta^3 + \varepsilon\sin\omega t;\end{aligned}$$

Для вариантов 21-30:

$$\begin{aligned}\dot{\theta} &= v, \\ \dot{v} &= -\sin^3\theta + \varepsilon\cos\omega t - \delta\dot{\theta};\end{aligned}$$

Построить функцию Мельникова для 1) $\alpha = 0.1, \gamma = 0.1, \omega = 1$;
2) $\alpha = 0.2, \gamma = 0.1, \omega = 1$;

3) $\alpha = 0.4, \gamma = 0.1, \omega = 1$;

4) $\alpha = 0.1, \gamma = 0.2, \omega = 1$;

5) $\alpha = 0.1, \gamma = 0.4, \omega = 1$;

6) $\alpha = 0.1, \gamma = 0.1, \omega = 2$;

7) $\alpha = 0.2, \gamma = 0.1, \omega = 2$;

8) $\alpha = 0.4, \gamma = 0.1, \omega = 2$;

9) $\alpha = 0.1, \gamma = 0.2, \omega = 1/\pi$;

10) $\alpha = 0.1, \gamma = 0.3, \omega = 2/\pi$;

11) $\alpha = 1, \varepsilon = 0.1, \omega = 1$;

12) $\alpha = 1, \varepsilon = 0.2, \omega = 1$;

13) $\alpha = 2, \varepsilon = 0.3, \omega = 1$;

14) $\alpha = 2, \varepsilon = 0.4, \omega = 1$;

15) $\alpha = 1, \varepsilon = 0.3, \omega = 2$;

16) $\alpha = 2, \varepsilon = 0.3, \omega = 2/\pi$;

17) $\alpha = 1, \varepsilon = 0.3, \omega = \pi^{-1}$;

18) $\alpha = 0.2, \varepsilon = 0.3, \omega = \pi^{-1}$;

19) $\alpha = 0.5, \varepsilon = 0.2, \omega = \pi^{-1}$;

20) $\alpha = 2, \varepsilon = 0.1, \omega = \pi^{-1}$;

21) $\varepsilon = 0.1, \delta = 0.1, \omega = 1$;

22) $\varepsilon = 0.2, \delta = 0.1, \omega = 1$;

23) $\varepsilon = 0.3, \delta = 0.2, \omega = 1$;

24) $\varepsilon = 0.4, \delta = 0.1, \omega = 1$;

25) $\varepsilon = -0.1, \delta = 0.2, \omega = 1$;

26) $\varepsilon = 0.1, \delta = -0.05, \omega = 1$;

27) $\varepsilon = 0.1, \delta = 0.2, \omega = 2$;

28) $\varepsilon = 0.1, \delta = 0.3, \omega = 2$;

29) $\varepsilon = 0.1, \delta = 0.1, \omega = \pi^{-1}$;

30) $\varepsilon = 0.3, \delta = 0.1, \omega = \pi^{-1}$.

Литература

1. Дьяконов, В.П. Maple 9.5, 10 в математике, физике и образовании[Текст]/В.П. Дьяков - М.:Солон-Пресс , 2006. - 720 с.
2. Голоскоков, Д. П. Уравнения математической физики. Решение задач в Maple[Текст]/Д.П. Голоскоков - Спб.:Питер , 2004. - 544 с.
3. Гукенхеймер, Дж. Нелинейные колебания, динамические системы и бифуркации векторных полей[Текст]/Дж. Гукенхеймер, Ф. Холмс. - Москва-Ижевск: Институт компьютерных исследований, 2002. - 560 с.
4. Боголюбов, Н.Н. Введение в нелинейную механику[Текст]/Н.Н. Боголюбов, Н.М. Крылов. - Москва-Ижевск: РХД , 2004. - 352 с.
5. Holmes, P.J. Bifurcation of the forced van der Pol oscillator [Text]/P.J. Holmes, D.A. Rand//Quart. Appl. Math.- 1978. - V. 35 - P. 495-509.